

## IDŹ DO

PRZYKŁADOWY ROZDZIAŁ



SPIS TREŚCI

## KATALOG KSIĄŻEK

KATALOG ONLINE

ZAMÓW DRUKOWANY KATALOG

## TWÓJ KOSZYK

DODAJ DO KOSZYKA

## CENNIK I INFORMACJE

ZAMÓW INFORMACJE  
O NOWOŚCIACH

ZAMÓW CENNIK

## CZYTELNIA

FRAGMENTY KSIĄŻEK ONLINE

# Excel. Najlepsze sztuczki i chwyt

Autor: John Walkenbach

Tłumaczenie: Łukasz Suma

ISBN: 83-246-0321-2

Tytuł oryginału: [John Walkenbachs](#)

[Favorite Excel Tips & Tricks](#)

Format: B5, stron: 400



### Excel naprawdę może pracować wydajniej!

- Przyspiesz proces wprowadzania danych
- Wykorzystaj formuły i funkcje
- Zautomatyzuj pracę za pomocą makr

Arkusze kalkulacyjne Excel to jeden z najpopularniejszych programów na świecie. Jest codziennie używany przez miliony ludzi, jednak większość z nich nie zna nawet połowy jego niesamowitych możliwości. Jeśli zajrzemy „pod maskę”, poznamy te cechy aplikacji, dzięki którym możemy pracować szybciej, wygodniej i efektywniej. Samodzielne odkrywanie możliwości programu to ciekawe zajęcie, jednak pochłania mnóstwo czasu. Poradnik, który je kompleksowo prezentuje, stanowi więc nieocenioną pomoc.

Książka „Excel. Najlepsze sztuczki i chwyt” zawiera ponad 200 wskazówek, dzięki którym nauczysz się optymalizować rutynowe procedury, budować dynamiczne wykresy i przetwarzać dane z wykorzystaniem formuł. Dowiesz się, jak rozwiązywać najczęstsze problemy związane z konfiguracją aplikacji, tworzyć własne dodatki, ukrywać przyciski pól na wykresach przestawnych, kontrolować automatyczne funkcje oraz rejestrować i uruchamiać makra w skrośszytach.

- Korzystanie ze skrótów klawiaturowych
- Zaznaczanie komórek
- Konfigurowanie interfejsu użytkownika
- Formatowanie danych i arkuszy
- Stosowanie formuł
- Tworzenie wykresów
- Drukowanie arkuszy
- Rejestrowanie i stosowanie makr
- Korzystanie z VBA

**Zostań mistrzem Excela**



# Spis treści

<b>O autorze .....</b>	<b>9</b>
<b>Wstęp .....</b>	<b>11</b>
<b>Rozdział 1. Podstawy korzystania z programu Excel .....</b>	<b>19</b>
Wersje programu Excel .....	21
Zwiększanie wydajności korzystania z menu programu .....	23
Wydajne zaznaczanie komórek .....	25
„Specjalne” zaznaczanie zakresów .....	29
Cofanie, ponowne wykonywanie i powtarzanie operacji .....	31
Zmiana liczby poziomów cofania operacji .....	32
Kilka przydatnych skrótów klawiaturowych .....	34
Przemieszczanie się pomiędzy arkuszami w ramach skoroszytu .....	35
Zerowanie znacznika używanego obszaru arkusza kalkulacyjnego .....	36
Różnica między skoroszytami i oknami .....	37
Unikanie używania okienka zadań podczas korzystania z systemu pomocy programu Excel 2003 .....	38
Dostosowywanie domyślnego skoroszytu .....	40
Zmiana wyglądu zakładki arkusza .....	42
Ukrywanie elementów interfejsu użytkownika .....	43
Ukrywanie kolumn i wierszy .....	46
Ukrywanie zawartości komórek .....	46
Przeprowadzanie niedokładnych wyszukiwań .....	47
Zmiana formatowania .....	49
Zwiększanie liczby wierszy i kolumn .....	51
Ograniczanie użytecznej powierzchni arkusza kalkulacyjnego .....	53
Używanie rozwiązań alternatywnych dla komentarzy do komórek .....	56
Zmiana rozmiaru tekstu w oknie systemu pomocy programu Excel .....	57
Skuteczne ukrywanie arkusza kalkulacyjnego .....	57
Rozwiązywanie typowych problemów z konfiguracją programu .....	59
<b>Rozdział 2. Wprowadzanie danych .....</b>	<b>65</b>
Wprowadzenie do typów danych .....	67
Przemieszczanie wskaźnika aktywnej komórki po wprowadzeniu danych .....	71
Zaznaczanie zakresu komórek wejściowych przed wprowadzaniem danych .....	72
Korzystanie z opcji Autouzupelnianie do automatyzacji wprowadzania danych .....	73
Zapewnianie wyświetlania nagłówek dzięki możliwości blokowania okienek .....	74
Automatyczne wypełnianie zakresu komórek arkusza z wykorzystaniem serii .....	75
Praca z ułamkami .....	77

Odczytywanie danych za pomocą narzędzia „Tekst na mowę” .....	79
Kontrolowanie automatycznych hiperłączy .....	80
Wprowadzanie numerów kart kredytowych .....	83
Używanie formularza wprowadzania danych oferowanego przez program Excel .....	83
Dostosowywanie i udostępnianie wpisów Autokorekty .....	85
Ograniczanie możliwości przemieszczania kursora jedynie do komórek wprowadzania danych .....	86
Kontrolowanie Schowka pakietu Office .....	88
Tworzenie listy rozwijanej w komórce arkusza .....	89
<b>Rozdział 3. Formatowanie .....</b>	<b>93</b>
Szybkie formatowanie liczb .....	95
Używanie odłączanych pasków narzędzi .....	95
Tworzenie niestandardowych formatów liczbowych .....	97
Używanie niestandardowych formatowań liczb do skalowania wartości .....	100
Używanie niestandardowych formatowań wartości daty i czasu .....	102
Kilka przydatnych niestandardowych formatowań liczbowych .....	103
Wyświetlanie tekstów i wartości liczbowych w jednej komórce .....	107
Scalanie komórek .....	109
Formatowanie poszczególnych znaków w komórce arkusza .....	110
Wyświetlanie wartości czasu większych niż 24 godziny .....	111
Przywracanie liczbom wartości numerycznych .....	112
Używanie funkcji Autoformatowanie .....	113
Posługiwanie się liniami siatki, obramowaniami oraz podkreśleniami .....	115
Tworzenie formatowań wykorzystujących efekty trójwymiarowe .....	117
Zawijanie tekstu w komórce .....	118
Przeglądanie wszystkich dostępnych znaków czcionki .....	119
Wprowadzanie znaków specjalnych .....	121
Używanie stylów nazwanych .....	122
Sposób obsługi kolorów przez program Excel .....	125
Stosowanie naprzemiennego wypełniania wierszy arkusza .....	127
Używanie obrazu graficznego w charakterze tła arkusza kalkulacyjnego .....	130
<b>Rozdział 4. Podstawowe formuły i funkcje .....</b>	<b>131</b>
Kiedy używać odwołań bezwzględnych .....	133
Kiedy używać odwołań mieszanych .....	134
Zmiana typu odwołań do komórek .....	135
Sztuczki z poleceniem Autosumowanie .....	136
Używanie statystycznych możliwości paska stanu .....	138
Konwertowanie formuł na wartości .....	139
Przetwarzanie danych bez korzystania z formuł .....	139
Przetwarzanie danych za pomocą formuł .....	140
Usuwanie wartości przy zachowaniu formuł .....	142
Używanie argumentów funkcji .....	143
Opisywanie formuł bez konieczności używania komentarzy .....	144
Tworzenie dokładnej kopii zakresu komórek przechowujących formuły .....	145
Kontrolowanie komórek z formułami z dowolnego miejsca arkusza kalkulacyjnego ...	146
Wyświetlanie i drukowanie formuł .....	147
Unikanie wyświetlania błędów w formułach .....	148
Używanie narzędzia Szukaj wyniku .....	150
Sekret związany z nazwami .....	152
Używanie nazwanych stałych .....	153
Używanie funkcji w nazwach .....	154

Edytowanie odwołań nazw .....	156
Używanie dynamicznych nazw .....	156
Tworzenie nazw na poziomie arkusza .....	158
Obsługa dat sprzed roku 1900 .....	159
Przetwarzanie ujemnych wartości czasu .....	161

## **Rozdział 5. Przydatne przykłady formuł ..... 163**

Wyznaczanie dat dni świątecznych .....	165
Obliczanie średniej ważonej .....	167
Obliczanie wieku osób .....	168
Szeregowanie wartości za pomocą formuły tablicowej .....	170
Zliczanie znaków w komórce .....	171
Wyrażanie liczb w postaci liczebników porządkowych w języku angielskim .....	172
Wyodrębnianie słów z tekstów .....	173
Rozdzielanie nazwisk .....	174
Usuwanie tytułów z nazwisk .....	176
Generowanie serii dat .....	176
Określanie specyficznych dat .....	178
Wyświetlanie kalendarza w zakresie komórek arkusza .....	181
Różne metody zaokrąglania liczb .....	182
Zaokrąglanie wartości czasu .....	185
Pobieranie zawartości ostatniej niepustej komórki w kolumnie lub wierszu .....	186
Używanie funkcji LICZ.JEŻELI .....	187
Zliczanie komórek spełniających wiele kryteriów jednocześnie .....	189
Obliczanie liczby różnych wpisów w zakresie .....	191
Obliczanie sum warunkowych wykorzystujących pojedynczy warunek .....	192
Obliczanie sum warunkowych wykorzystujących wiele warunków .....	194
Wyszukiwanie wartości dokładnej .....	196
Przeprowadzanie wyszukiwań dwuwymiarowych .....	198
Przeprowadzanie wyszukiwania w dwóch kolumnach .....	200
Przeprowadzanie wyszukiwania przy użyciu tablicy .....	201
Używanie funkcji ADR.POŚR .....	202
Tworzenie megaformuł .....	204

## **Rozdział 6. Wykresy i elementy grafiki ..... 207**

Tworzenie wykresu tekstowego bezpośrednio w zakresie komórek .....	209
Komentowanie zawartości wykresu .....	211
Tworzenie samopowiększającego się wykresu .....	212
Tworzenie kombinacji wykresów .....	214
Obsługa brakujących danych na wykresie liniowym .....	216
Tworzenie wykresów Ganta .....	218
Tworzenie wykresów przypominających termometr .....	219
Tworzenie wykresów wykorzystujących elementy graficzne .....	222
Wykreślanie matematycznych funkcji jednej zmiennej .....	224
Wykreślanie matematycznych funkcji dwóch zmiennych .....	225
Tworzenie półprzezroczystych serii danych na wykresie .....	227
Zapisywanie wykresu w postaci pliku graficznego .....	228
Ustalanie identycznych rozmiarów wykresów .....	230
Wyświetlanie wielu wykresów w jednym arkuszu wykresu .....	232
„Zamrażanie” wykresu .....	232
Dodawanie „znaku wodnego” do arkusza .....	235
Zmiana kształtu pola komentarza do komórki .....	236
Wstawianie grafiki w pole komentarza do komórki .....	237

<b>Rozdział 7. Analiza danych i listy .....</b>	<b>239</b>
Korzystanie z możliwości związanych z listami w Excelu 2003 .....	241
Sortowanie w porządku określonym dla więcej niż trzech kolumn .....	243
Używanie widoków niestandardowych wraz z możliwościami automatycznego filtrowania .....	244
Umieszczanie wyników działania zaawansowanego filtra w różnych arkuszach kalkulacyjnych .....	246
Porównywanie dwóch zakresów za pomocą formatowania warunkowego .....	247
Układanie rekordów listy w przypadkowej kolejności .....	249
Wypełnianie pustych miejsc w raporcie .....	251
Tworzenie listy z tabeli podsumowania .....	253
Odnajdowanie powtórzeń przy użyciu formatowania warunkowego .....	255
Uniemożliwianie wstawiania wierszy lub kolumn w ramach zakresu .....	257
Szybkie tworzenie tabeli liczby wystąpień .....	259
Kontrolowanie odwołań do komórek w tabeli przestawnej .....	261
Grupowanie elementów w tabeli przestawnej według dat .....	262
Ukrywanie przycisków pól na wykresie przestawnym .....	264
<b>Rozdział 8. Praca z plikami .....</b>	<b>267</b>
Importowanie pliku tekstowego do zakresu komórek arkusza .....	269
Pobieranie danych ze strony WWW .....	270
Wyświetlanie pełnej ścieżki dostępu do skoroszytu .....	274
Zapisywanie podglądu skoroszytu .....	275
Korzystanie z właściwości dokumentu .....	276
Sprawdzanie informacji o użytkowniku, który otworzył plik jako ostatni .....	278
Odszukiwanie brakującego przycisku „Nie na wszystkie” podczas zamykania plików .....	280
Pobieranie listy nazw plików .....	281
Znaczenie haseł programu Excel .....	283
Używanie plików obszaru roboczego .....	283
Zmniejszanie rozmiaru skoroszytu .....	284
<b>Rozdział 9. Drukowanie .....</b>	<b>285</b>
Wybieranie elementów do wydrukowania .....	287
Umieszczanie powtarzających się wierszy lub kolumn na wydruku .....	288
Drukowanie nieciągłych zakresów komórek na jednej stronie .....	289
Uniemożliwianie drukowania obiektów .....	291
Sztuczki związane z numerowaniem stron .....	292
Podgląd podziału stron .....	294
Dodawanie i usuwanie znaków podziału stron .....	296
Drukowanie danych do pliku PDF .....	297
Unikanie drukowania określonych wierszy .....	297
Drukowanie arkusza na jednej stronie .....	300
Drukowanie formuł .....	301
Kopiowanie ustawień strony pomiędzy arkuszami .....	303
Używanie widoków niestandardowych przy drukowaniu .....	304
<b>Rozdział 10. Dostosowywanie menu i pasków narzędzi .....</b>	<b>307</b>
Odnajdowanie wielofunkcyjnych przycisków pasków narzędzi .....	309
Odszukiwanie ukrytych poleceń menu .....	310
Dostosowywanie menu i pasków narzędzi .....	310
Tworzenie niestandardowego paska narzędzi .....	312
Wyłączanie automatycznego wyświetlania pasków narzędzi .....	314
Dołączanie pasków narzędzi do arkuszy kalkulacyjnych .....	315
Tworzenie kopii zapasowych samodzielnie dostosowanych menu i pasków narzędzi .....	316

<b>Rozdział 11. Znajdowanie, naprawianie i unikanie błędów .....</b>	<b>317</b>
Korzystanie z możliwości sprawdzania błędów w Excelu .....	319
Znajdowanie komórek formuł .....	321
Metody radzenia sobie z problemami związanymi z liczbami zmiennoprzecinkowymi .....	323
Tworzenie tabeli nazw komórek i zakresów .....	324
Graficzne przeglądanie nazw .....	325
Odszukiwanie „ślepych” łącz .....	325
Różnica między wartościami wyświetlanymi a rzeczywistymi .....	326
Śledzenie powiązań występujących pomiędzy komórkami .....	327
<b>Rozdział 12. Podstawy języka VBA i korzystanie z makr .....</b>	<b>331</b>
Podstawowe informacje o makrach i języku VBA .....	333
Rejestrowanie makra .....	334
Zagadnienia bezpieczeństwa związane z makrami .....	337
Korzystanie ze skoroszytu makr osobistych .....	339
Różnice między funkcjami a procedurami .....	340
Wyświetlanie okien komunikatów .....	342
Pobieranie informacji od użytkownika .....	345
Uruchamianie makra przy otwieraniu skoroszytu .....	346
Tworzenie prostych funkcji arkusza kalkulacyjnego .....	349
Sprawianie, by Excel przemówił .....	351
Ograniczenia funkcji niestandardowych .....	352
Wywoływanie poleceń menu za pomocą makra .....	353
Zapisywanie funkcji niestandardowych w postaci dodatku do programu .....	354
Wyświetlanie połączonej kontrolki kalendarza .....	355
Używanie dodatków do programu Excel .....	357
<b>Rozdział 13. Konwersje i obliczenia matematyczne .....</b>	<b>361</b>
Przeliczanie wartości między różnymi systemami jednostek .....	363
Konwersja temperatur .....	366
Wyznaczanie parametrów trójkątów prostokątnych .....	367
Obliczanie pól powierzchni, obwodów oraz pojemności .....	369
Rozwiązywanie liniowych układów równań .....	372
Generowanie unikalnych całkowitych liczb losowych .....	373
Generowanie liczb losowych .....	375
Obliczanie pierwiastków i reszt z dzielenia .....	376
Obliczanie średniej warunkowej .....	377
<b>Rozdział 14. Źródła informacji na temat programu Excel .....</b>	<b>379</b>
Używanie systemu pomocy programu Excel .....	381
Wyszukiwanie pomocy w internecie .....	382
Korzystanie z grup dyskusyjnych dotyczących Excela .....	382
Ciekawe strony WWW na temat Excela .....	383
<b>Skorowidz .....</b>	<b>387</b>

Rozdział 5.

# **Przydatne przykłady formuł**

W rozdziale tym znajdziesz wiele przykładów formuł. Niektóre z nich będziesz mógł wykorzystać dokładnie w takiej formie, w jakiej zostały przedstawione. Inne zaś będziesz musiał dostosować do swoich własnych potrzeb.

# Spis sposobów

<b>Sposób 85.</b>	Wyznaczanie dat dni świątecznych	165
<b>Sposób 86.</b>	Obliczanie średniej ważonej	167
<b>Sposób 87.</b>	Obliczanie wieku osób	168
<b>Sposób 88.</b>	Szeregowanie wartości za pomocą formuły tablicowej	170
<b>Sposób 89.</b>	Zliczanie znaków w komórce	171
<b>Sposób 90.</b>	Wyrażanie liczb w postaci liczebników porządkowych w języku angielskim	172
<b>Sposób 91.</b>	Wyodrębnianie słów z tekstów	173
<b>Sposób 92.</b>	Rozdzielanie nazwisk	174
<b>Sposób 93.</b>	Usuwanie tytułów z nazwisk	176
<b>Sposób 94.</b>	Generowanie serii dat	176
<b>Sposób 95.</b>	Określanie specyficznych dat	178
<b>Sposób 96.</b>	Wyświetlanie kalendarza w zakresie komórek arkusza	181
<b>Sposób 97.</b>	Różne metody zaokrąglania liczb	182
<b>Sposób 98.</b>	Zaokrąglanie wartości czasu	185
<b>Sposób 99.</b>	Pobieranie zawartości ostatniej niepustej komórki w kolumnie lub wierszu	186
<b>Sposób 100.</b>	Używanie funkcji LICZ.JEŻELI	187
<b>Sposób 101.</b>	Zliczanie komórek spełniających wiele kryteriów jednocześnie	189
<b>Sposób 102.</b>	Obliczanie liczby różnych wpisów w zakresie	191
<b>Sposób 103.</b>	Obliczanie sum warunkowych wykorzystujących pojedynczy warunek	192
<b>Sposób 104.</b>	Obliczanie sum warunkowych wykorzystujących wiele warunków	194
<b>Sposób 105.</b>	Wyszukiwanie wartości dokładnej	196
<b>Sposób 106.</b>	Przeprowadzanie wyszukiwań dwuwymiarowych	198
<b>Sposób 107.</b>	Przeprowadzanie wyszukiwania w dwóch kolumnach	200
<b>Sposób 108.</b>	Przeprowadzanie wyszukiwania przy użyciu tablicy	201
<b>Sposób 109.</b>	Używanie funkcji ADR.POŚR	202
<b>Sposób 110.</b>	Tworzenie megaformuł	204



## Sposób 85. Wyznaczanie dat dni świątecznych

Określenie dat niektórych świąt może być dość skomplikowane. Część z nich jest bardziej niż oczywista, ponieważ zawsze występują w tych samych dniach każdego roku, żeby wymienić tu chociażby Nowy Rok czy Święto Niepodległości. W przypadku tego typu świąt powinieneś po prostu skorzystać z funkcji DATA. Aby na przykład sprawdzić, jakim dniem tygodnia będzie Nowy Rok (który zawsze przypada 1 stycznia) roku określonego za pomocą danej znajdującej się w komórce A1, należy sformatować odpowiednio komórkę (na przykład dddd — zajrzyj do sposobu 42.) i skorzystać z następującej formuły:

```
=DATA(A1;1;1)
```

Inne święta są zdefiniowane jako określone wystąpienie pewnego dnia tygodnia w konkretnym miesiącu lub są wręcz uzależnione od faz księżyca. Przykładem może tu być większość obchodzonych w Polsce świąt kościelnych, takich jak Boże Ciało czy Wielkanoc, bądź też niektóre z państwowych świąt amerykańskich, takich jak Dzień Prezydenta czy też Święto Dziękczynienia.

Przy tworzeniu wszystkich wymienionych poniżej formuł założono, że wartość określająca rok znajduje się w komórce A1. Zwróć uwagę na fakt, że ponieważ Nowy Rok, Święto Wojska Polskiego, Święto Niepodległości czy Boże Narodzenie obchodzone są zawsze tego samego dnia roku, obliczenie ich dat sprowadza się do prostego wywołania funkcji DATA.

### Nowy Rok

Święto to zawsze przypada dnia 1 stycznia, więc odpowiednia dla niego formuła będzie miała postać:

```
=DATA(A1;1;1)
```

### Dzień Martina Luthera Kinga

To amerykańskie święto wypada w trzeci poniedziałek stycznia. Przedstawiona poniżej formuła oblicza datę święta Martina Luthera Kinga w roku określonym zawartością komórki A1:

```
=DATA(A1;1;1)+JEŻELI(2<DZIEŃ.TYG(DATA(A1;1;1));7-DZIEŃ.TYG(DATA(A1;1;1))  
+2;2-DZIEŃ.TYG(DATA(A1;1;1)))+((3-1)*7)
```

### Dzień Prezydenta

Dzień ten w Stanach Zjednoczonych jest wyznaczony na trzeci poniedziałek lutego. Datę tę w roku zdefiniowanym w komórce A1 można obliczyć, korzystając z następującej formuły:

```
=DATA(A1;2;1)+JEŻELI(2<DZIEŃ.TYG(DATA(A1;2;1));7-DZIEŃ.TYG(DATA(A1;2;1))
+2;2-DZIEŃ.TYG(DATA(A1;2;1)))+((3-1)*7)
```

## Wielkanoc

Wyznaczenie daty Wielkanocy jest dość trudne z uwagi na sposób określenia dnia tego święta. Jest to bowiem pierwsza niedziela po pierwszej pełni księżyca występującej po równonocy wiosennej, która przypada 21 marca. Przedstawię tu formułę znalazłem w internecie i szczerze mówiąc, nie mam pojęcia, w jaki sposób działa:

```
=ZAOKR.W.DÓŁ(DATA(A1;5;DZIEŃ(MINUTA(A1/38)/2+56));7)-34
```

Pamiętaj tu o wybraniu dla komórki któregoś z formatów daty, gdyż w innym przypadku zostanie w niej wyświetlona niewiele znacząca wartość numeryczna.

Powyższa formuła zwraca poprawną datę Niedzieli Wielkanocnej dla lat z przedziału od roku 1900 do 2078. Myślę, że ten zakres okaże się wystarczający dla większości użytkowników programu. Jeśli w Twoim przypadku jest inaczej, będziesz mógł poszukać odpowiedniego rozwiązania w sieci. Na stronach internetowych znaleźć można liczne kody makr VBA pozwalających na wyznaczenie daty Wielkanocy na wiele różnych sposobów.

## Święto Konstytucji 3 Maja

Działanie jest tu proste, gdyż — jak sama nazwa wskazuje — święto to przypada zawsze dnia 3 maja:

```
=DATA(A1;5;3)
```

## Dzień Pamięci

W ostatni poniedziałek maja Amerykanie obchodzą Dzień Pamięci. Formuła pozwalająca obliczyć datę tego dnia w roku podanym w komórce *A1* ma postać:

```
=DATA(A1;6;1)+JEŻELI(2<DZIEŃ.TYG(DATA(A1;6;1));7-DZIEŃ.TYG(DATA(A1;6;1))
+2;2-DZIEŃ.TYG(DATA(A1;6;1)))+((1-1)*7)-7
```

Zwróć uwagę, że powyższa formuła oblicza tak naprawdę datę pierwszego poniedziałku czerwca określonego roku, a następnie odejmuje liczbę 7 w celu wyznaczenia poniedziałku o tydzień wcześniejszego, czyli ostatniego w maju.

## Święto Pracy

Święto Pracy obchodzone jest w Stanach Zjednoczonych zupełnie innego dnia niż w Europie, gdyż wypada ono w pierwszy poniedziałek września. Formuła wyznaczająca tę datę dla roku określonego w komórce *A1* ma następującą postać:

```
=DATA(A1;9;1)+JEŻELI(2<DZIEŃ.TYG(DATA(A1;9;1));7-DZIEŃ.TYG(DATA(A1;9;1))  
+2;2-DZIEŃ.TYG(DATA(A1;9;1)))+((1-1)*7)
```

Oczywiście, żeby wyznaczyć dzień Święta Pracy w Polsce, zastosujesz raczej dużo prostszą formułę o postaci:

```
=DATA(A1;5;1)
```

## Dzień Krzysztofa Kolumba

To amerykańskie święto przypada na drugi poniedziałek października. Poniższa formuła pozwala na wyznaczenie jego daty w roku określonym w komórce *A1*:

```
=DATA(A1;10;1)+JEŻELI(2<DZIEŃ.TYG(DATA(A1;10;1));7-DZIEŃ.TYG(DATA(A1;10;1))  
+2;2-DZIEŃ.TYG(DATA(A1;10;1)))+((2-1)*7)
```

## Święto Niepodległości

Święto to ustalono na dzień 11 listopada:

```
=DATA(A1;11;11)
```

## Święto Dziękczynienia

Jedno z najważniejszych świąt w Stanach Zjednoczonych obchodzone jest w czwartą czwartek listopada. Datę Święta Dziękczynienia w roku podanym w komórce *A1* można obliczyć przy użyciu następującej formuły:

```
=DATA(A1;11;1)+JEŻELI(5<DZIEŃ.TYG(DATA(A1;11;1));7-DZIEŃ.TYG(DATA(A1;11;1))  
+5;5-DZIEŃ.TYG(DATA(A1;11;1)))+((4-1)*7)
```

## Boże Narodzenie

Jak wiadomo, święto to przypada na dzień 25 grudnia:

```
=DATA(A1;12;25)
```

# Sposób 86. Obliczanie średniej ważonej

Oferowana przez program Excel funkcja ŚREDNIA zwraca średnią (czy też przeciętną) wartość liczb znajdujących się w określonym zakresie komórek. Bardzo często jednak zachodzi konieczność obliczenia średniej ważonej. Możesz stracić na poszukiwania całej

dzień, lecz mimo to nie znajdziesz funkcji Excela, która przeprowadzałaby podobne działanie. Masz jednak możliwość obliczenia średniej ważonej za pomocą odpowiedniej formuły używającej funkcji SUMA.ILOCZYNÓW oraz SUMA.

Na rysunku 86.1 przedstawiono prosty przykład arkusza kalkulacyjnego zawierającego ceny paliwa gazowego odnotowane w okresie 30 dni. Na przykład przez pierwsze cztery dni miesiąca litr gazu kosztował 2,48 zł, jego cena spadła następnie do poziomu 2,41 zł i utrzymała tę wartość przez dwa kolejne dni, by potem znów zmaleć na kolejne trzy dni do kwoty 2,39 zł i tak dalej.

### Rysunek 86.1.

Formuła znajdujaca się w komórce B16 oblicza średnią ważoną cen płynnego gazu

	A	B	C	D
1	<b>Ceny gazu w styczniu</b>			
2				
3		<b>Cena</b>	<b>Ilość dni</b>	
4		2,48	4	
5		2,41	2	
6		2,39	3	
7		2,35	1	
8		2,34	1	
9		2,41	2	
10		2,37	2	
11		2,42	4	
12		2,41	3	
13		2,47	8	
14				
15	Średnia:	2,405		
16	Średnia ważona:	2,428		
17				
18				

W komórce B15 umieszczono formułę, która używa funkcji ŚREDNIA:

```
=ŚREDNIA(B4:B13)
```

Ale, wbrew temu, co może się wydawać, formuła ta nie zwraca właściwego wyniku. Aby takowy otrzymać, poszczególnym cenom musiałyby być przypisane odpowiednie wagi związane z ilością dni, przez które obowiązywała każda z wartości. Innymi słowy, właściwym sposobem obliczania wartości średniej byłaby tu raczej średnia ważona.

Średnią taką można obliczyć za pomocą poniższej formuły, która w arkuszu została umieszczona w komórce B16:

```
=SUMA.ILOCZYNÓW(B4:B13;C4:C13)/SUMA(C4:C13)
```

## Sposób 87. Obliczanie wieku osób

Obliczanie wieku ludzkiego w programie Excel wymaga użycia pewnej sztuczki, ponieważ wynik nie zależy wyłącznie od bieżącego roku, lecz również od aktualnego dnia, a sytuację komplikują dodatkowo lata przestępne.

Przedstawię tutaj trzy metody obliczania wieku osób. W wykorzystanych do tego celu formułach przyjąłem założenie, że data urodzenia znajduje się w komórce *B1*, zaś w komórce *B2* umieszczona jest aktualna data, tak jak zostało to pokazane na rysunku 87.1.

### Rysunek 87.1.

Obliczanie  
wieku osób

	A	B	C
1	Data urodzenia:	1973-05-19	
2	Dziś:	2005-12-29	
3	Wiek:	32	
4			
5			
6			

## Metoda 1

Poniższa formuła odejmuje datę urodzenia od aktualnej daty i dzieli otrzymany wynik przez liczbę 365,25. Funkcja ZAOKR.DO.CAŁK usuwa część ułamkową rezultatu.

```
=ZAOKR.DO.CAŁK((B2-B1)/365,25)
```

Formuła ta nie jest dokładna w stu procentach, ponieważ przeprowadza dzielenie przez średnią liczbę dni w roku. W niektórych przypadkach zatem zwraca niepoprawne wyniki. Przykładem może tu być obliczanie wieku dziecka, które ma dokładnie rok; w sytuacji takiej powyższa formuła zwróci wartość 0 zamiast 1.

## Metoda 2

Bardziej dokładną metodą obliczania wieku będzie zastosowanie funkcji YEARFRAC, która jest dostępna w ramach dodatku *Analysis ToolPak*.

```
=ZAOKR.DO.CAŁK(YEARFRAC(B2;B1))
```

## Metoda 3

Trzecia metoda wyznaczania wieku korzysta z funkcji DATA.RÓŻNICA. W zależności od tego, której wersji Excela aktualnie używasz, może się zdarzyć, że funkcja ta nie będzie udokumentowana w systemie pomocy programu.

```
=DATA.RÓŻNICA(B1;B2;"Y")
```

Jeśli bardzo zależy Ci na dokładności, możesz zastosować nieco zmodyfikowaną wersję tej formuły:

```
=DATA.RÓŻNICA(B1;B2;"Y") & " lat, " & DATA.RÓŻNICA(B1;B2;"YM") & " miesięcy, " & DATA.RÓŻNICA(B1;B2;"MD") & " dni"
```

Formuła ta zwróci ciąg znaków podobny do przedstawionego poniżej:

```
32 lat, 7 miesięcy, 10 dni
```

## Sposób 88. Szeregowanie wartości za pomocą formuły tablicowej

Ustalanie porządku wartości znajdujących się w zakresie komórek okazuje się czasem bardzo przydatną możliwością. Jeśli masz na przykład arkusz kalkulacyjny zawierający dane o rocznych wartościach sprzedaży osiągniętych przez dwudziestu przedstawicieli handlowych Twojej firmy, możesz dzięki temu dokonać klasyfikacji każdej z tych osób i dowiedzieć się, kto zajmuje jaką pozycję w rankingu sprzedaży przedsiębiorstwa, zaczynając od najwyższej, a kończąc na najniższej.

Jeżeli zdarzyło Ci się już korzystać z oferowanej przez program Excel funkcji POZYCJA, z pewnością zauważyłeś, że wyniki jej działania nie zawsze w pełni Ci odpowiadają. Jeśli bowiem dwie wartości mają zajmować na przykład trzecie miejsce, funkcja POZYCJA obydwu przypisze pozycję 3, a Ty być może wolałbyś przypisać im jakąś wartość średnią czy też środkową, co w tym przypadku oznaczałoby pozycję 3,5 dla obu danych.

Na rysunku 88.1 przedstawiono arkusz kalkulacyjny, w którym zastosowano obie wymienione wyżej metody pozycjonowania kolumny wartości. Pierwsza z tych metod — wyniki jej działania widoczne są w kolumnie C — korzysta ze standardowej funkcji POZYCJA programu Excel. W kolumnie D natomiast umieszczono wyniki działania formuł tablicowych zastosowanych do ustalenia pozycji poszczególnych liczb na liście. Zakres komórek B2:B9 nosi nazwę *WartośćSprzedaży*.

### Rysunek 88.1.

*Ustalanie pozycji danych za pomocą funkcji POZYCJA oferowanej przez program Excel oraz przy wykorzystaniu odpowiedniej formuły tablicowej*

	A	B	C	D	E
	Handlowiec	Wartość sprzedaży	Pozycja ustalona przez funkcję Excela	Pozycja ustalona z wyk. formuły tablicowej	
1	Adamski	45,000	7	7	
2	Bigielski	90,000	3	3,5	
3	Fryderak	90,000	3	3,5	
4	Grzegórzka	54,000	6	6	
5	Janiczek	25,000	8	8	
6	Jedluga	101,000	1	1	
7	Marcinkiewicz	85,000	5	5	
8	Szumski	92,000	2	2	
9					
10					
11					

Poniżej znajduje się formuła umieszczona w komórce D2; skopiowano ją również do komórek widocznych pod nią:

$$=SUMA(1*(B2<=WartośćSprzedaży))-(SUMA(1*(B2=WartośćSprzedaży))-1)/2$$


Formuła tablicowa jest szczególnym rodzajem formuły i działa w odniesieniu do danych umieszczonych w tablicy. Podczas wprowadzania takiej formuły powinieneś nacisnąć kombinację klawiszy *Ctrl+Shift+Enter* zamiast samego klawisza *Enter*, aby powiadomić program, że wpisana została właśnie formuła tablicowa, nie zaś zwykła. Excel wyświetla tego typu formuły w nawiasach klamrowych, co ma na celu przypomnienie Ci, że masz do czynienia z formułą tablicową. Efektu tego nie osiągniesz, wpisując nawiasy klamrowe z klawiatury.

## Sposób 89. Zliczanie znaków w komórce

Tutaj znajdziesz przykłady formuł, których zadaniem jest obliczanie liczby znaków wpisanych do komórki.

### Zliczanie wystąpień określonych znaków w komórce

Podana niżej formuła oblicza liczbę wystąpień litery B (tylko wielkiej litery) w ciągu znaków umieszczonym w komórce *A1*:

```
=DŁ(A1)-DŁ(PODSTAW(A1;"B";""))
```

Działanie tej formuły opiera się na wykorzystaniu funkcji `PODSTAW` tworzącej w pamięci programu nowy ciąg znaków, z którego usunięto wszystkie litery B. Kolejnym krokiem jest odjęcie długości otrzymanego ciągu od długości oryginalnego tekstu znajdującego się w komórce i uzyskanie w ten sposób informacji na temat liczby wystąpień w nim litery B.

Jeśli w komórce *A1* będzie się na przykład znajdował tekst *Biały bąbłowiec*, formuła zwróci wartość 1.

Przedstawiona poniżej formuła jest bardziej uniwersalna, gdyż pozwala na obliczenie liczby wystąpień litery B — zarówno wielkiej, jak i małej — w tekście znajdującym się w komórce *A1*:

```
=DŁ(A1)-DŁ(PODSTAW(LITERY.WIELKIE(A1);"B";""))
```

Umieszczenie w komórce *A1* tekstu *Biały bąbłowiec* spowoduje, że formuła zwróci wartość 3.

### Zliczanie wystąpień ciągu znakowego w komórce

Kolejna przedstawiona tu formuła pozwala na znajdowanie liczby wystąpień konkretnego ciągu znaków. Zwraca ona liczbę wystąpień określonego ciągu tekstowego znajdującego się w komórce *B1* w tekście umieszczonym w komórce *A1*. Poszukiwany ciąg tekstowy może zawierać dowolną liczbę znaków.

```
=(DŁ(A1)-DŁ(PODSTAW(A1;B1;"")))/DŁ(B1)
```

Jeśli na przykład w komórce *A1* zostanie umieszczony tekst *Leśnik na Leśniku*, a w *B2* będzie się znajdował ciąg znaków *Leśnik*, powyższa formuła zwróci liczbę 2.

Wykonywane porównanie uwzględnia wielkość zastosowanych liter, a więc umieszczenie w komórce *B1* tekstu *leśnik* spowoduje, że formuła zwróci wartość 0. Aby ominąć

to ograniczenie, powinieneś skorzystać ze zmodyfikowanej wersji formuły, która ma następującą postać:

```
=(DŁ(A1)-DŁ(PODSTAW(LITERY.WIELKIE(A1); LITERY.WIELKIE(B1);"")))/DŁ(B1)
```

## Sposób 90. Wyrażanie liczb w postaci liczebników porządkowych w języku angielskim

Przydatna bywa nieraz możliwość wyrażania liczb w postaci liczebników porządkowych. Zamienianie liczb na pełne słowa byłoby, co prawda, czynnością zbyt skomplikowaną, a tworzenie skrótów cyfrowo-literowych w takim przypadku jest w języku polskim niepoprawne, w odróżnieniu od języka angielskiego, gdzie jest to normalną praktyką. Liczba 21 traktowana jako liczebnik porządkowy jest w nim na przykład wyrażana poprzez dodanie odpowiedniej końcówki, którą w tym przypadku jest *st* — a więc liczebnik przyjmuje postać *21st*. Program Excel nie oferuje specjalnego formatu liczbowego, który pomógłby w takiej sytuacji, możliwe jest jednak opracowanie odpowiedniej formuły, która wypełni to zadanie.

W języku angielskim istnieją cztery końcówki dodawane do liczby w celu uzyskania liczebnika porządkowego. Są to: *st*, *nd*, *rd* i *th*. Wybór jednej z nich zależy od wartości przekształcanej liczby, a rządząca nim reguła jest dość zawiła. Z tego powodu odpowiednia formuła również będzie dość skomplikowana. Większość liczb wymaga użycia końcówki *th*. Wyjątkami od tej reguły będą liczby kończące się cyframi 1, 2 i 3, jednak nie takie, których drugą od końca cyfrą jest 1, a więc nie wartości kończące się liczbami 11, 12 i 13. Zasada ta może się wydawać dość zagmatwana, ale da się ją przełożyć na język zrozumiały dla Excela, a więc na formułę.

Przedstawiona poniżej formuła przekształca liczbę całkowitą umieszczoną w komórce *A1* na odpowiedni liczebnik porządkowy języka angielskiego:

```
=A1&JEŻELI(LUB(WARTOŚĆ(PRAWY(A1;2))={11;12;13});"th";JEŻELI(LUB(WARTOŚĆ(PRAWY(A1))={1;2;3});WYBIERZ(PRAWY(A1);"st";"nd";"rd");"th"))
```

Formuła ta jest dość skomplikowana, postaram się więc wytłumaczyć Ci jej sposób działania. Jest on w skrócie taki:

1. Jeśli ostatnie dwie cyfry liczby to 11, 12 lub 13, użyj końcówki *th*.
2. Jeśli zasada 1. nie znajduje zastosowania, sprawdź ostatnią cyfrę. Jeżeli ostatnią cyfrą liczby jest 1, użyj końcówki *st*. Jeżeli ostatnią cyfrą liczby jest 2, skorzystaj z końcówki *nd*. Jeżeli ostatnią cyfrą liczby jest 3, użyj końcówki *rd*.
3. Jeśli żadna z powyższych zasad nie została zastosowana, użyj końcówki *th*.

Na rysunku 90.1 przedstawiono efekty działania podanej wyżej formuły.



**Rysunek 90.1.**

*Korzystanie z formuły do wyrażania liczb w postaci angielskich liczebników porządkowych*

	A	B	C
1	Liczba	Liczebnik porządkowy	
2	1	1st	
3	4	4th	
4	7	7th	
5	10	10th	
6	13	13th	
7	16	16th	
8	19	19th	
9	22	22nd	
10	25	25th	
11	28	28th	
12	31	31st	
13	34	34th	
14	37	37th	
15	40	40th	
16	43	43rd	
17	46	46th	
18	49	49th	
19	52	52nd	
20			
21			

## Sposób 91. Wyodrębnianie słów z tekstów

Zaprezentowane tutaj formuły będą przydatne do wyodrębniania słów z ciągów znaków znajdujących się w komórkach arkusza kalkulacyjnego. Jednej z nich możesz na przykład użyć do wydzielenia pierwszego słowa z tekstu.

### Wyodrębnianie pierwszego słowa z ciągu tekstowego

Aby wydobyć pierwsze słowo z określonego tekstu, formuła musi zlokalizować w nim pozycję pierwszego znaku spacji, a następnie użyć tej informacji jako argumentu funkcji LEWY. Działanie takie wykonuje następująca formuła:

```
=LEWY(A1;ZNAJDŹ(" ";A1)-1)
```

Zwraca ona wszystkie znaki, które znajdują się w tekście umieszczonym w komórce *A1* przed wystąpieniem pierwszej spacji. Pojawia się tu jednak pewien problem — jeśli w komórce tej nie występuje żaden znak spacji, bo zawiera ona tylko jedno słowo, formuła zwróci kod błędu. Nieco bardziej rozbudowana wersja formuły rozwiązuje ten kłopot dzięki wykorzystaniu dodatkowych funkcji JEŻELI oraz CZY.BŁ do sprawdzenia faktu wystąpienia błędu:

```
=JEŻELI(CZY.BŁ(ZNAJDŹ(" ";A1));A1;LEWY(A1;ZNAJDŹ(" ";A1)-1))
```

## Wyodrębnianie ostatniego słowa z ciągu tekstowego

Wydobycie ostatniego słowa z łańcucha tekstowego jest nieco trudniejsze, ponieważ funkcja ZNAJDŹ przeszukuje teksty zawsze od lewej do prawej strony. Z tego powodu problemem jest tu znalezienie **ostatniego** znaku spacji w zadanym ciągu. Istnieje jednak pewne rozwiązanie, czego najlepszym dowodem jest zaprezentowana poniżej formuła. Zwraca ona ostatnie słowo należące do tekstu, czyli wszystkie znaki znajdujące się po ostatniej spacji, która w nim występuje:

```
=PRAWY(A1;DŁ(A1)-ZNAJDŹ("*";PODSTAW(A1;" ";"*";DŁ(A1)-DŁ(PODSTAW(A1;" ";""))))
```

Z formułą tą wiąże się jednak ten sam problem, który pojawił się w przypadku pierwszej formuły przedstawionej wyżej: zwraca ona kod błędu w sytuacji, gdy zadany ciąg znaków nie zawiera przynajmniej jednej spacji. Zmodyfikowana wersja formuły wykorzystuje funkcję JEŻELI do sprawdzenia, czy w tekście umieszczonym w komórce *A1* znajdują się jakiegokolwiek znaki spacji. Jeśli ich nie ma, zwrócona zostanie cała zawartość tej komórki. W innym przypadku do akcji wkroczy przedstawiona wcześniej formuła:

```
=JEŻELI(CZY.BŁ(ZNAJDŹ(" ";A1));A1;PRAWY(A1;DŁ(A1)-ZNAJDŹ("*";PODSTAW(A1;" ";"*";DŁ(A1)-DŁ(PODSTAW(A1;" ";""))))))
```

## Wyodrębnianie wszystkich słów z wyjątkiem pierwszego z ciągu tekstowego

Następująca formuła zwraca zawartość komórki *A1* z pominięciem pierwszego słowa:

```
=PRAWY(A1;DŁ(A1)-ZNAJDŹ(" ";A1;1))
```

Jeśli komórka *A1* będzie zawierała tekst Wstępny budżet na rok 2006, powyższa formuła zwróci ciąg znaków budżet na rok 2006.

Formuła ta zwróci natomiast kod błędu, gdy w komórce będzie się znajdować tylko jedno słowo. Problem ten rozwiązano w przedstawionej niżej formule, która w podobnej sytuacji zwróci pusty ciąg tekstowy:

```
=JEŻELI(CZY.BŁ(ZNAJDŹ(" ";A1));"";PRAWY(A1;DŁ(A1)-ZNAJDŹ(" ";A1;1)))
```

## Sposób 92. Rozdzielanie nazwisk

Załóżmy, że masz listę pełnych imion i nazwisk ludzi, znajdującą się w jednej kolumnie. Twoim zadaniem jest rozdzielenie tych nazwisk na trzy kolumny w taki sposób, aby w pierwszej z nich znalazły się pierwsze imiona, w kolejnej drugie imiona lub inicjały, zaś w trzeciej nazwiska. Zadanie to jest bardziej skomplikowane, niż mogłoby się początkowo wydawać, ponieważ nie we wszystkich nazwiskach występujących w kolumnie użyto drugich imion czy też dodatkowych inicjałów. Mimo to problem jest możliwy do rozwiązania.



Opisane powyżej zadanie będzie **dużo** bardziej skomplikowane, gdy na liście znajdują się jeszcze nazwiska poprzedzone tytułami, takimi jak Pan czy dr, lub nazwiska zawierające dodatkowe informacje, w rodzaju Jr. czy III. Przedstawione tu rozwiązania **nie** uwzględniają tego typu trudnych przypadków, mimo to wyniki ich działania nadal stanowią będą dobry punkt wyjścia, a z pojedynczymi błędnymi wpisami będziesz mógł sobie poradzić, ręcznie edytując odpowiednie komórki.

We wszystkich zaprezentowanych niżej formułach przyjęto założenie, że imiona i nazwisko umieszczone są w komórce *A1*.

W prosty sposób możesz opracować formułę, która będzie wyodrębniała imię:

```
=LEWY(A1;ZNAJDŹ(" ";A1)-1)
```

Poniższa formuła będzie natomiast zwracała nazwisko:

```
=PRAWY(A1;DŁ(A1)-ZNAJDŹ("*";PODSTAW(A1;" ";"*";DŁ(A1)-DŁ(PODSTAW(A1;" ";""))))
```

Następująca formuła wydobywa z całości zapisu drugie imię. Przy jej tworzeniu założono, że pierwsze imię znajduje się w komórce *B1*, a wyodrębnione nazwisko umieszczone zostało w komórce *D1*:

```
=JEŻELI(DŁ(B1&D1)+2>=DŁ(A1);" ";FRAGMENT.TEKSTU(A1;DŁ(B1)+2;DŁ(A1)-DŁ(B1&D1)-2))
```

Jak możesz zauważyć na rysunku 92.1, przedstawione tu formuły spisują się całkiem nieźle. W widocznym na nim arkuszu występują, co prawda, pewne problemy, zwłaszcza w przypadku obcych nazwisk szlacheckich, w których pojawiają się dodatkowe słowa typu Van, ale zwykle nazwiska rozdzielane są poprawnie. Poza tym, jak już wcześniej wspominałem, te nieliczne błędy możesz poprawić ręcznie.

### Rysunek 92.1.

*W arkuszu tym użyto formuł do wyodrębnienia pierwszego imienia, drugiego imienia lub jego inicjału oraz nazwiska z wpisów imion i nazwisk znajdujących się na liście widocznej w kolumnie A*

	A	B	C	D	E
1	<b>Imię i nazwisko</b>	<b>Imię</b>	<b>Drugie imię</b>	<b>Nazwisko</b>	
2	Jan M. Rokita	Jan	M.	Rokita	
3	Luiza Królowa	Luiza		Królowa	
4	A.L. Kołacz	A.L.		Kołacz	
5	A. Bibek	A.		Bibek	
6	Robert W. Burczyk	Robert	W.	Burczyk	
7	Janusz A. Van Klompf	Janusz	A. Van	Klompf	
8	Pan Tadeusz G. Szula	Pan	Tadeusz G.	Szula	
9					
10					



W wielu przypadkach będziesz mógł wyeliminować konieczność używania formuł dzięki oferowanemu Ci przez program poleceniu *Dane/Tekst jako kolumny...*. Pozwala ono na rozdzielanie tekstu na poszczególne elementy składowe. Wybranie tej komendy spowoduje wywołanie okna dialogowego *Kreator konwersji tekstu na kolumny*, który w kilku krokach przeprowadzi Cię przez proces przetwarzania pojedynczej kolumny danych w zbiór kolumn. W pierwszym kroku działania kreatora będziesz przeważnie używał opcji *Rozdzielany*, a w drugim kroku jako ogranicznik tekstu wybierzesz spację.

## Sposób 93. Usuwanie tytułów z nazwisk

Może się zdarzyć sytuacja, w której będziesz zmuszony do usunięcia tytułów (takich jak Pan, Pani czy Państwo) poprzedzających nazwiska znajdujące się na liście umieszczonej w arkuszu Excela. Operację tę będziesz prawdopodobnie chciał przeprowadzić przed opisanym wcześniej rozdzielaniem pełnych nazwisk na ich części składowe.

Z zamieszczonej poniżej formuły będziesz mógł skorzystać w celu usunięcia z komórek przechowujących nazwiska trzech występujących najczęściej tytułów, czyli słów Pan, Pani oraz Państwo. Jeśli komórka *A1* będzie na przykład zawierała nazwisko Pan Fryderyk Misiasty, efektem działania formuły będzie ciąg znaków Fryderyk Misiasty.

```
=JEŻELI(LUB(LEWY(A1;4)="Pan ";LEWY(A1;5)="Pani ";LEWY(A1;8)="Państwo ");
PRAWY(A1;DŁ(A1)-ZNAJDŹ(" ";A1));A1)
```

W powyższej formule sprawdzane są trzy warunki. Jeśli zechcesz sprawdzać większą ich liczbę, na przykład w celu wyeliminowania kolejnych tytułów, powinieneś po prostu dodać odpowiednie argumenty w wywołaniu funkcji LUB.

## Sposób 94. Generowanie serii dat

Z pewnością często zdarza się, że chcesz wprowadzić do arkusza serię dat. Na przykład przy zapisywaniu tygodniowych wartości obrotów firmy będziesz chciał wprowadzić serię dat oddzielonych od siebie o siedem dni. Daty te mogą służyć do identyfikowania liczb opisujących sprzedaż.

### Używanie możliwości Autowypełnienie


Najbardziej efektywna metoda wprowadzania serii danych nie wykorzystuje jakichkolwiek formuł, sprowadza się bowiem do użycia możliwości automatycznego wypełniania kolejnych komórek arkusza następującymi po sobie datami. Żeby z niej skorzystać, powinieneś wpisać pierwszą datę, a następnie przeciągnąć uchwyt wypełniania komórek przy użyciu prawego przycisku myszki. Po zwolnieniu przycisku na ekranie pojawi się menu kontekstowe, z którego będziesz mógł wybrać odpowiednią dla siebie opcję, tak jak zostało to przedstawione na rysunku 94.1.

### Używanie formuł

Przewagą rozwiązania wykorzystującego formuły nad używaniem funkcji *Autowypełnienie* do utworzenia serii dat jest możliwość zmiany pierwszej daty w serii, co pociągnie za sobą aktualizację wszystkich pozostałych danych. W celu skorzystania z tego

**Rysunek 94.1.**

*Korzystanie z możliwości Autowypełnienie w celu utworzenia serii dat*



rozwiązania powinienś jedynie wpisać do pierwszej komórki właściwą datę początkową, a następnie do kolejnych komórek wprowadzić formuły, których zadaniem będzie generowanie odpowiednich wartości.

Przy tworzeniu przedstawionych niżej przykładów formuł przyjęto założenie, że pierwsza data serii umieszczona została w komórce *A1*, a pierwsza formuła znajduje się w komórce *A2*. Odpowiednią liczbę kolejnych komórek należy po prostu wypełnić kopią tej formuły.

Aby otrzymać serię dat oddzielonych od siebie okresem siedmiu dni, użyj następującej formuły:

```
=A1+7
```

Aby wygenerować serię dat odległych od siebie o miesiąc, skorzystaj z formuły:

```
=DATA(ROK(A1);MIESIĄC(A1)+1;DZIEŃ(A1))
```

W celu otrzymania serii dat odległych od siebie dokładnie o rok zastosuj poniższą formułę:

```
=DATA(ROK(A1)+1;MIESIĄC(A1);DZIEŃ(A1))
```

Aby wygenerować serię dat składających się wyłącznie z dni tygodnia (bez sobót i niedziel), powinienś skorzystać z zamieszczonej poniżej formuły. Formuła utworzona została przy założeniu, że data znajdująca się w komórce *A1* jest dniem powszednim, czyli nie jest sobotą lub niedzielą.

```
=JEŻELI(DZIEŃ.TYG(A1)=6;A1+3;A1+1)
```

## Sposób 95. Określanie specyficznych dat

Tutaj znajdziesz szereg przydatnych formuł, które zwracają pewne specyficzne daty.

### Określanie dnia roku

1 stycznia jest pierwszym dniem każdego roku, a 31 grudnia jest jego dniem ostatnim. Ale co z pozostałymi dniami, znajdującymi się pomiędzy nimi? Przedstawiona poniżej formuła zwraca kolejny numer dnia w roku dla daty przechowywanej w komórce *A1*:

```
=A1-DATA(ROK(A1);1;0)
```

Kolejny numer dnia w roku określany jest czasem mianem **daty juliańskiej**.

Następująca formuła zwraca liczbę dni, które pozostały do końca roku, licząc od podanej daty umieszczonej w komórce *A1*:

```
=DATA(ROK(A1);12;31)-A1
```



Wprowadzenie którejkolwiek z powyższych formuł spowoduje, że program Excel zastosuje formatowanie wartości daty w przypadku przechowywujących je komórek. Będziesz więc musiał sformatować je za pomocą któregoś z formatów numerycznych, aby móc przeglądać wyniki działania formuł w postaci liczbowej.

### Określanie dnia tygodnia

Jeśli zajdzie potrzeba wyznaczenia, na jaki dzień tygodnia przypada określona data, z pomocą przyjdzie Ci funkcja **DZIEŃ.TYG.** Funkcja ta przyjmuje argument stanowiący datę i zwraca liczbę całkowitą z przedziału od 1 do 7, która odpowiada numerowi dnia w tygodniu — przy założeniu, że tydzień zaczyna się w niedzielę. Podana niżej formuła zwraca na przykład wartość 1, gdyż pierwszym dniem roku 2006 jest właśnie niedziela:

```
=DZIEŃ.TYG(DATA(2006;1;1))
```



Dzień tygodnia dla określonej daty możesz również wyznaczyć, stosując do przechowywanej ją komórki odpowiednie formatowanie niestandardowe. Aby dzień tygodnia wyświetlany był w postaci słowa stanowiącego jego nazwę, powinieneś na przykład zastosować następujący ciąg formatujący:

```
dddd
```

Pamiętaj jednak, że komórka naprawdę będzie nadal przechowywała pełną datę, a nie jedynie kolejny numer dnia tygodnia, jak w przypadku rozwiązania korzystającego z formuły.

Funkcja DZIEŃ.TYG umożliwia również podanie drugiego, opcjonalnego argumentu, który określa stosowany przez nią system numerowania dni w tygodniu. Jeśli użyjesz w tym celu liczby 2, funkcja zwróci wartość 1 dla poniedziałku, 2 dla wtorku i tak dalej. Zastosowanie liczby 3 jako drugiego argumentu funkcji DZIEŃ.TYG spowoduje, że w przypadku poniedziałku zwrócona zostanie wartość 0, w przypadku wtorku — 1 i tak dalej.

## Określanie daty ostatniej niedzieli

Formuła, którą tu przedstawiam, zwraca datę ostatniego wystąpienia określonego dnia tygodnia. Możesz z niej skorzystać na przykład do wyznaczenia daty ostatniej niedzieli, przy czym, jeśli aktualnym dniem jest niedziela, formuła zwróci datę dzisiejszą. Pamiętaj o takim sformatowaniu komórki, aby wyświetlane były wartości daty.

```
=DZIS()-MOD(DZIS()-1;7)
```

Aby zmodyfikować powyższą formułę w celu wyznaczania dat innych dni niż niedziela, powinieneś zmienić występującą w niej liczbę 1 na inną wartość z przedziału od 2 (w przypadku poniedziałku) do 7 (dla soboty).

## Określanie pierwszego dnia tygodnia występującego po podanej dacie

Znajdująca się poniżej formuła może być wykorzystana do wyznaczenia daty podanego dnia tygodnia, który będzie następował po określonej dacie. Możesz więc dzięki niej na przykład sprawdzić datę, jaką będzie miał pierwszy poniedziałek po 1 czerwca 2006 roku.

Przy tworzeniu formuły przyjęto założenie, że w komórce *A1* znajduje się data, a komórka *A2* zawiera liczbę z przedziału od 1 do 7, która określa dzień tygodnia, przy czym 1 oznacza niedzielę, 2 — poniedziałek i tak dalej.

```
=A1+A2-DZIEŃ.TYG(A1)+(A2<DZIEŃ.TYG(A1))*7
```

W przypadku, gdy w komórce *A1* znajduje się data 1 czerwiec 2006, a komórka *A2* zawiera oznaczającą poniedziałek liczbę 2, wynikiem działania formuły będzie 5 czerwiec 2006 — ten dzień bowiem przypada w pierwszy poniedziałek po 1 czerwca 2006 (czwartek).

## Określanie n-tego wystąpienia podanego dnia tygodnia w miesiącu

Podczas Twojej pracy może Ci się czasem przydać formuła pozwalająca na wyznaczenie daty określonego wystąpienia w danym miesiącu pewnego dnia tygodnia. Wyobraź sobie na przykład, że dniem wypłaty pensji w Twojej firmie jest zawsze drugi piątek miesiąca, a Twoim zadaniem jest określenie dat wszystkich wypłat w rozpoczynającym się właśnie roku. Odpowiednie obliczenia wykona dla Ciebie poniższa formuła:

```
=DATA(A1;A2;1)+A3-DZIEŃ.TYG(DATA(A1;A2;1))+(A4-(A3>=DZIEŃ.TYG(DATA(A1;A2;1))))*7
```

Przy tworzeniu tej formuły przyjęte zostały następujące założenia:

- ♦ komórka *A1* zawiera rok,
- ♦ komórka *A2* przechowuje miesiąc,
- ♦ w komórce *A3* umieszczony jest kolejny numer odpowiedniego dnia tygodnia, czyli liczba 1 dla niedzieli, 2 dla poniedziałku i tak dalej,
- ♦ komórka *A4* zawiera numer poszukiwanego wystąpienia określonego dnia, czyli na przykład 2 w sytuacji, gdy chcesz wyznaczyć datę drugiego wystąpienia dnia tygodnia określonego argumentem przechowywanym w komórce *A3*.

Wykorzystanie tej formuły do określenia daty pierwszego piątku czerwca 2006 spowoduje otrzymanie wartości 2 czerwiec 2006.

## Określanie ostatniego dnia miesiąca

W celu znalezienia daty odpowiadającej ostatniemu dniu określonego miesiąca możesz skorzystać z funkcji `DATA`. Należy tu wykorzystać fakt, że zerowy dzień następnego miesiąca jest traktowany przez tę funkcję jako ostatni dzień miesiąca go poprzedzającego, a więc trzeba podać jej argument miesiąca zwiększony o 1 i wartość dnia równą 0.

Przy opracowywaniu podanej niżej formuły założono, że w komórce *A1* znajduje się data określająca wybrany miesiąc. Formuła w wyniku swojego działania zwróci ostatni dzień tego właśnie miesiąca.

```
=DATA(ROK(A1);MIESIĄC(A1)+1;0)
```

Wariacji tej formuły możesz użyć do wyznaczenia liczby dni wchodzących w skład podanego miesiąca. Zaprezentowana poniżej formuła zwraca liczbę całkowitą określającą liczbę dni miesiąca zdefiniowanego za pomocą daty umieszczonej w komórce *A1*. Upewnij się, że komórka przechowująca tę formułę korzysta ze zwykłego formatowania liczbowego, nie zaś z formatowania daty.

```
=DZIEŃ(DATA(ROK(A1);MIESIĄC(A1)+1;0))
```

## Określanie kwartału, do którego należy podany dzień

Przy tworzeniu raportów finansowych pomocna może się okazać możliwość prezentowania informacji odnoszących się do poszczególnych kwartałów danego roku. Podana niżej formuła zwraca liczbę całkowitą z przedziału od 1 do 4. Liczba ta określa kwartał, do którego należy data znajdująca się w komórce *A1*:

```
=ZAOKR.GÓRA(MIESIĄC(A1)/3;0)
```

Działanie tej formuły polega na podzieleniu numeru miesiąca przez liczbę 3, a następnie zaokrągleniu otrzymanego wyniku w górę.




## Sposób 96. Wyświetlanie kalendarza w zakresie komórek arkusza

Tutaj znajdziesz opis metody tworzenia w zakresie komórek dynamicznego kalendarza na dowolny miesiąc wybranego roku. Na rysunku 96.1 przedstawiono przykładowy kalendarz tego typu. Zmiana daty widocznej w jego górnej części spowoduje, że kalendarz zostanie przeliczony od nowa tak, aby wyświetlane były daty dla podanego roku i miesiąca.

### Rysunek 96.1.

*Pokazany tu kalendarz został utworzony za pomocą skomplikowanej formuły tablicowej*



	A	B	C	D	E	F	G	H	I	
1										
2		marzec 2006								
3		Nie	Pon	Wto	Sro	Czw	Pią	Sob		
4					1	2	3	4		
5		5	6	7	8	9	10	11		
6		12	13	14	15	16	17	18		
7		19	20	21	22	23	24	25		
8		26	27	28	29	30	31			
9										
10										
11										

Aby utworzyć ten kalendarz w komórkach **B2:H9**, postępuj według następujących instrukcji:

1. Zaznacz zakres komórek **B2:H2** i scal komórki, klikając przycisk *Scal i wyśrodkuj* widoczny na pasku narzędzi *Formatowanie*.
2. Do scalonego zakresu wprowadź datę. Podany dzień miesiąca nie będzie tu miał żadnego znaczenia.
3. Do komórek **B3:H3** wpisz skróty nazw dni tygodnia.
4. Zaznacz zakres komórek **B4:H9**, a następnie wpisz podaną niżej formułę. Pamiętaj, że jest to formuła tablicowa, więc aby ją wprowadzić, zamiast klawisza *Enter* będziesz musiał na koniec nacisnąć kombinację klawiszy *Ctrl+Shift+Enter*.

```
=JEŻELI(MIESIĄC(DATA(ROK(B2);MIESIĄC(B2);1))<>MIESIĄC(DATA(ROK(B2);
MIESIĄC(B2);1)-(DZIEŃ.TYG(DATA(ROK(B2);MIESIĄC(B2);1))-1)+{0\1\2\3\4\5}
*7+{1;2;3;4;5;6;7}-1);"";DATA(ROK(B2);MIESIĄC(B2);1)-(DZIEŃ.TYG(DATA(
ROK(B2);MIESIĄC(B2);1))-1)+{0\1\2\3\4\5}*7+{1;2;3;4;5;6;7}-1)
```

5. Sformatuj zakres komórek **B4:H9**, korzystając z niestandardowego formatu daty w taki sposób, aby wyświetlane były tylko dni. Ciąg formatujący będzie tu miał postać: **d**.
6. Dostosuj odpowiednio szerokość kolumn i dobierz wszelkie inne niezbędne formatowania komórek.

Zmiana miesiąca i roku w dacie widocznej w górnej części zakresu spowoduje, że kalendarz zostanie automatycznie zaktualizowany. Po utworzeniu kalendarza będziesz mógł skopiować przechowujący go zakres komórek i wstawić do każdego innego arkusza kalkulacyjnego i skoroszytu.

## Sposób 97. Różne metody zaokrąglania liczb

Zaokrąglanie liczb jest bardzo typową czynnością przeprowadzaną w programie Excel, dlatego też znajdziesz w nim szereg funkcji umożliwiających wykonanie tego zadania na różne sposoby.

Bardzo ważną rzeczą jest tu właściwe zrozumienie różnicy między zaokrąglaniem wartości a ich formatowaniem. Gdy **formatujesz** liczbę w taki sposób, aby wyświetlana była z określoną ilością miejsc dziesiętnych, formuły korzystające z liczby będą używać jej rzeczywistej wartości, która może się różnić od tego, co widać w komórce arkusza. Gdy **zaokrąglasz** liczbę, używające jej formuły będą stosować tę właśnie zaokrągloną wartość.

W tabeli 97.1 przedstawione zostały funkcje oferowane przez program Excel, opracowane z myślą o zaokrągleniu wartości.

**Tabela 97.1.** Funkcje Excela służące do zaokrąglania liczb

Funkcja	Opis
ZAOKR.W.GÓRĘ	Zaokrągla liczbę w górę (czyli w kierunku od zera) do najbliższej wielokrotności określonej liczby.
DOLLARDE*	Zmienia cenę wyrażoną w postaci ułamkowej na wartość w postaci dziesiętnej.
DOLLARFR*	Zmienia cenę wyrażoną w postaci dziesiętnej na wartość w postaci ułamkowej.
ZAOKR.DO.PARZ	Zaokrągla liczby dodatnie w górę (w kierunku od zera), a liczby ujemne w dół (również w kierunku od zera) do najbliższej całkowitej liczby parzystej.
ZAOKR.W.DÓŁ	Zaokrągla liczbę w dół (czyli w kierunku do zera) do najbliższej wielokrotności określonej liczby.
ZAOKR.DO.CAŁK	Zaokrągla liczbę w dół do najbliższej jej wartości całkowitej.
MROUND*	Zaokrągla liczbę do wielokrotności określonej liczby.
ZAOKR.DO.NPARZ	Zaokrągla liczby dodatnie w górę (w kierunku od zera), a liczby ujemne w dół (również w kierunku od zera) do najbliższej całkowitej liczby nieparzystej.
ZAOKR	Zaokrągla liczbę do podanej liczby cyfr po przecinku.
ZAOKR.DÓŁ	Zaokrągla liczbę w dół (w kierunku do zera) do podanej liczby cyfr po przecinku.
ZAOKR.GÓRA	Zaokrągla liczbę w górę (w kierunku od zera) do podanej liczby cyfr po przecinku.
LICZBA.CAŁK	W swoim domyślnym działaniu obcina liczbę do wartości całkowitej, usuwając przy tym ewentualną część ułamkową. Opcjonalny drugi argument steruje dokładnością obcinania.

\* Funkcje te stają się dostępne po zainstalowaniu dodatku *Analysis ToolPak*.

Podane w dalszej części niniejszego sposobu przykłady formuł pozwolą Ci lepiej zrozumieć działanie różnych metod zaokrąglania liczb.

## Zaokrąglanie do najbliższej wielokrotności określonej liczby

Funkcja `MROUND`, będąca częścią dodatku *Analysis ToolPak*, przydaje się do zaokrąglania wartości do najbliższej wielokrotności określonej liczby. Możesz jej więc na przykład użyć w celu zaokrąglenia liczby 133 do najbliższej wielokrotności liczby 5. Zamieszczona poniżej formuła zwróci wówczas wartość 135:

```
=MROUND(133;5)
```

Podobny efekt można też uzyskać za pomocą standardowej funkcji Excela `ZAOKR.W.GÓRĘ`:

```
=ZAOKR.W.GÓRĘ(133;5)
```

Otrzymany wynik będzie tu taki sam, jak w przypadku użycia funkcji `MROUND`. Jednak wywołanie wspomnianych funkcji z pierwszym argumentem równym 131 da odmienne wyniki. Funkcja `ZAOKR.W.GÓRĘ` ponownie zwróci wartość 135, ale `MROUND` zwróci 130. Dzieje się tak dlatego, że `MROUND` szuka najbliższej wielokrotności podanej liczby, zaokrąglając w górę lub w dół, a `ZAOKR.W.GÓRĘ`, jak sama nazwa wskazuje, zawsze zaokrągla w górę.

Istnieje jeszcze funkcja `ZAOKR.W.DÓŁ`, która powoduje zaokrąglenie w dół pierwszego argumentu do najbliższej wielokrotności liczby określonej drugim argumentem. Zarówno przy wartości pierwszego argumentu równym 131, jak i 133 oraz wielokrotności równej 5 formuła zwróci wynik 130.

## Zaokrąglanie wartości walutowych

Bardzo często zdarzają się sytuacje, w których konieczne jest zaokrąglenie wartości walutowych. Nagle okazuje się, że obliczona cena jakiegoś produktu wynosi na przykład 45,78923 zł. W takiej sytuacji z pewnością będziesz chciał zaokrąglić otrzymaną wartość do najbliższego grosza. Może się to wydawać bardzo łatwe, jednak tak się składa, że działanie takie można przeprowadzić na trzy różne sposoby:

- ◆ zaokrąglić kwotę w górę do najbliższego grosza,
- ◆ zaokrąglić kwotę w dół do najbliższego grosza,
- ◆ zaokrąglić kwotę do najbliższego grosza w górę lub w dół.

Przedstawiona niżej formuła opracowana została przy założeniu, że wartość ceny wyrażona w złotówkach i groszach umieszczona jest w komórce `A1`. Zadaniem formuły jest zaokrąglenie tej wartości do najbliższego grosza. Jeśli zatem komórka `A1` będzie zawierać liczbę 12,421 zł, formuła zwróci wartość 12,42 zł, a w przypadku wartości 12,429 zwróci 12,43.

```
=ZAOKR(A1;2)
```

Jeżeli chcesz, aby wartości były zaokrąglane w górę do najbliższego grosza, powinieneś skorzystać z funkcji ZAOKR.W.GÓRĘ. Znajdująca się poniżej formuła używa jej do zaokrąglenia w ten sposób liczby umieszczonej w komórce A1. Wstawienie do tej komórki wartości 12,421 zł spowoduje, że formuła zwróci liczbę 12,43 zł.

```
=ZAOKR.W.GÓRĘ(A1;0,01)
```

Jeśli Twoim zadaniem jest zaokrąglanie wartości walutowych w dół, rozwiązaniem będzie użycie funkcji ZAOKR.W.DÓŁ. Na przykład zamieszczona poniżej formuła pozwala zaokrąglić w dół liczbę przechowywaną w komórce A1 w taki sposób, że wartość 12,421 zł zostanie przetworzona na 12,42 zł.

```
=ZAOKR.W.DÓŁ(A1;0,01)
```

Aby zaokrąglić w górę wartość oznaczającą kwotę wyrażoną w złotówkach do najbliższych pięciu groszy, powinieneś skorzystać z następującej formuły:

```
=ZAOKR.W.GÓRĘ(A1;0,05)
```

## Używanie funkcji ZAOKR.DO.CAŁK i LICZBA.CAŁK

Na pozór funkcje ZAOKR.DO.CAŁK i LICZBA.CAŁK wydają się niemal identyczne. Obie konwertują dowolną wartość liczbową do postaci liczby całkowitej. Różnica polega jednak na tym, że funkcja LICZBA.CAŁK po prostu obcina ułamkową część oryginalnej wartości, zaś funkcja ZAOKR.DO.CAŁK zaokrągla tę wartość do najbliższej liczby całkowitej w oparciu o ułamkową część pierwotnej liczby.

W praktyce różnica ta staje się widoczna przy przetwarzaniu liczb ujemnych. Na przykład poniższa formuła zwróci w wyniku swojego działania wartość -14,0:

```
=LICZBA.CAŁK(-14,2)
```

Kolejna zaś zwróci liczbę -15,0, ponieważ wartość -14,2 zostanie zaokrąglona w dół do najbliższej mniejszej od niej liczby całkowitej:

```
=ZAOKR.DO.CAŁK(-14,2)
```

Funkcja LICZBA.CAŁK umożliwia podanie dodatkowego (opcjonalnego) argumentu, który przydaje się przy przycinaniu ułamków dziesiętnych. Na przykład przedstawiona niżej formuła zwraca liczbę 54,33, czyli wartość przyciętą do dwóch miejsc po przecinku:

```
=LICZBA.CAŁK(54,333333;2)
```

## Zaokrąglanie do n cyfr znaczących

W niektórych przypadkach może Ci się bardzo przydać możliwość zaokrąglania wartości numerycznych do określonej liczby cyfr znaczących. Możesz na przykład chcieć wyrazić liczbę 1 432 187 za pomocą dwóch cyfr znaczących, co oznaczać będzie zamienienie jej na wartość 1 400 000. Z kolei wartość 9 187 877 przedstawiona przy użyciu trzech cyfr znaczących przyjmie postać 9 180 000.

Jeśli masz do czynienia z całkowitymi liczbami dodatnimi, przedstawiona poniżej formuła załatwia sprawę. Zaokrągła ona liczbę przechowywaną w komórce *A1* do dwóch cyfr znaczących. Jeżeli jednak będziesz chciał zaokrąglać wartości, używając innej liczby miejsc znaczących, powinieneś zastąpić występującą w niej liczbę 2 odpowiednią wartością.

```
=ZAOKR.DÓŁ(A1;2-DŁ(A1))
```

W przypadku liczb ujemnych i wartości niebędących liczbami całkowitymi rozwiązanie jest nieco bardziej skomplikowane. Zamieszczona niżej formuła stanowi bardziej uniwersalny sposób zaokrąglenia wartości znajdującej się w komórce *A1* do liczby cyfr znaczących zapisanej w komórce *A2*. Formuła ta przetwarza poprawnie zarówno całkowite, jak i niecałkowite liczby dodatnie i ujemne.

```
=ZAOKR(A1:A2-1-ZAOKR.DO.CAŁK(LOG10(MODUŁ.LICZBY(A1))))
```

Na przykład, jeśli w komórce *A1* będzie znajdować się liczba 1,27845, a w komórce *A2* wartość 3, formuła zwróci liczbę 1,28000, czyli wartość zaokrągloną do trzech cyfr znaczących.

## Sposób 98. Zaokrąglenie wartości czasu

Niewykluczone, że przydarzy Ci się sytuacja, w której będziesz musiał opracować formułę zaokrąglającą wartości czasu do określonej liczby minut. Możesz na przykład być zmuszony do wprowadzania zapisów dotyczących czasu pracy Twojej firmy z dokładnością do 15 minut. Tutaj przedstawię Ci kilka różnych metod zaokrąglenia wartości czasu.

Następująca formuła zaokrągli daną o czasie przechowywaną w komórce *A1* do najbliższej pełnej minuty:

```
=ZAOKR(A1*1440;0)/1440
```

Działanie formuły polega na przemnożeniu wartości czasu przez liczbę 1440 w celu otrzymania całkowitej liczby minut w jednej dobie, a następnie zaokrągleniu jej za pomocą funkcji ZAOKR i podzieleniu uzyskanego wyniku przez wartość 1440. Na przykład wpisanie do komórki *A1* czasu 11:52:34 spowoduje, że formuła zwróci wartość 11:53:00.

Kolejna formuła przypomina powyższą, z wyjątkiem tego, że zaokrągli wartość czasu przechowywaną w komórce *A1* do najbliższej pełnej godziny:

```
=ZAOKR(A1*24;0)/24
```

Jeśli komórka *A1* będzie zawierać wartość 5:21:31, wynikiem działania formuły będzie czas 5:00:00.

Zamieszczona poniżej formuła powoduje z kolei zaokrąglenie wartości znajdującej się w komórce *A1* do najbliższych 15 minut, czyli kwadransa:

```
=ZAOKR(A1*24/0,25;0)*(0,25/24)
```

W formule tej liczba 0,25 reprezentuje ułamek godziny. Aby zaokrąglić wartość czasu do najbliższych 30 minut, należy zastąpić liczbę 0,25 wartością 0,5, tak jak zrobiono to w poniższej formule:

```
=ZAOKR(A1*24/0,5;0)*(0,5/24)
```

## Sposób 99. Pobieranie zawartości ostatniej niepustej komórki w kolumnie lub wierszu

Załóżmy, że masz pewien arkusz kalkulacyjny, który często aktualizujesz, dodając nowe dane do jego kolumn. Może Ci się przydać w takiej sytuacji jakaś metoda odwoływania się do ostatniej wartości umieszczonej w określonej kolumnie — czyli, innymi słowy, do ostatnio wprowadzonej danej.

Na rysunku 99.1 przedstawiono przykład. Nowe dane są wpisywane codziennie, a Twoim zadaniem będzie tu opracowanie formuły, która zwraca ostatnią wartość znajdującą się w kolumnie *C*.

### Rysunek 99.1.

*Do pobierania zawartości ostatniej niepustej komórki w kolumnie C możesz użyć formuły*

	A	B	C	D
1	Data	Kwota	Bieżąca suma	
2	2005-06-01	604	604	
3	2005-06-02	532	1136	
4	2005-06-03	314	1450	
5	2005-06-04	873	2323	
6	2005-06-05	704	3027	
7	2005-06-06	532	3559	
8	2005-06-07	763	4322	
9	2005-06-08	983	5305	
10	2005-06-09			
11	2005-06-10			
12	2005-06-11			
13	2005-06-12			
14	2005-06-13			
15	2005-06-14			
16	2005-06-15			
17				
18				

Jeśli kolumna *C* nie zawiera pustych komórek, rozwiązanie jest dość proste:

```
=PRZESUNIĘCIE(C1;ILE.NIEPUSTYCH(C:C)-1;0)
```

Formuła ta korzysta z funkcji ILE.NIEPUSTYCH do obliczenia liczby niepustych komórek należących do kolumny *C*. Informacja ta, po odjęciu 1, jest następnie używana

w charakterze argumentu dla funkcji PRZESUNIĘCIE. Jeśli zatem na przykład ostatnia wartość jest umieszczona w wierszu 100, funkcja ILE.NIEPUSTYCH zwróci liczbę 100, zaś funkcja PRZESUNIĘCIE poda wartość umieszczoną w komórce znajdującej się 99 wierszy pod komórką C1 w tej samej kolumnie.

Jeżeli w kolumnie D znajduje się pewna liczba pustych komórek rozsianych w jej różnych miejscach, co zdarza się bardzo często, zaprezentowana powyżej formuła nie będzie spełniać swojego zadania w odniesieniu do tej kolumny, ponieważ funkcja ILE.NIEPUSTYCH nie liczy komórek pustych. Zadaniu temu jest w stanie sprostać przedstawiona niżej formuła tablicowa, która zwraca zawartość ostatniej niepustej komórki z pierwszych pięciuset wierszy kolumny D:

```
=INDEKS(D1:D500;MAX(WIERSZ(D1:D500)*(D1:D500<>"")))
```



Aby wprowadzić formułę tablicową, powinieneś nacisnąć kombinację klawiszy *Ctrl+Shift+Enter* zamiast samego klawisza *Enter*.

Możesz, oczywiście, w taki sposób zmodyfikować podaną wyżej formułę, aby jej działanie dotyczyło innej kolumny niż D. Aby to zrobić, powinieneś odpowiednio zmienić wszystkie sześć odwołań do niej widocznych w treści formuły. Jeśli ostatnia niepusta komórka może się pojawić poniżej wiersza 500, powinieneś, rzecz jasna, zmienić dwa wystąpienia liczby 500 na jakąś większą wartość. Pamiętaj jednak, że im mniejsza będzie liczba wierszy, do których odwołuje się formuła, tym większa będzie szybkość jej działania.

Zamieszczona niżej formuła tablicowa jest podobna do zaprezentowanej powyżej, ale zwraca zawartość ostatniej niepustej komórki podanego wiersza (w tym konkretnym przykładzie jest to wiersz 1):

```
=INDEKS(1:1;MAX(NR.KOLUMNY(1:1)*(1:1<>"")))
```

Aby wykorzystać tę formułę do przeszukiwania innego wiersza, powinieneś zmienić odwołanie 1:1 na odwołanie odpowiadające numerowi Twojego wiersza.

## Sposób 100. Używanie funkcji LICZ.JEŻELI

Oferowane przez program Excel funkcje ILE.LICZB i ILE.NIEPUSTYCH doskonale sprawdzają się w przypadku prostych operacji zliczania, jednak czasami będą Ci potrzebne nieco większe możliwości. Tutaj znajdziesz szereg przykładów formuł prezentujących potężne możliwości funkcji LICZ.JEŻELI, która pozwala na zliczanie komórek w oparciu o różnego rodzaju kryteria.

Wszystkie te formuły przeprowadzają swoje działania na zbiorze danych umieszczonym w zakresie o nazwie *Dane*. Podczas przeglądania tabeli 100.1 zauważysz zapewne,

że argument określający warunek uwzględnienia komórki w zliczaniu może być definiowany bardzo swobodnie. Możesz tu bowiem skorzystać ze stałych, wyrażeń, funkcji, odwołań do komórek, a nawet znaków globalnych (\* i ?).

**Tabela 100.1.** Przykłady formuł wykorzystujących funkcję LICZ.JEŻELI

Formuła	Działanie
=LICZ.JEŻELI(Dane;12)	Zwraca liczbę komórek zawierających wartość 12.
=LICZ.JEŻELI(Dane;"<0")	Zwraca liczbę komórek zawierających wartości ujemne.
=LICZ.JEŻELI(Dane;"<>0")	Zwraca liczbę komórek zawierających wartości różne od 0, przy czym komórki puste nie oznaczają wartości 0.
=LICZ.JEŻELI(Dane;">5")	Zwraca liczbę komórek zawierających wartości większe od liczby 5.
=LICZ.JEŻELI(Dane;A1)	Zwraca liczbę komórek zawierających wartości równe danej umieszczonej w komórce A1.
=LICZ.JEŻELI(Dane;">"&A1)	Zwraca liczbę komórek zawierających wartości większe od danej przechowywanej w komórce A1.
=LICZ.JEŻELI(Dane;"*")	Zwraca liczbę komórek zawierających wartości tekstowe.
=LICZ.JEŻELI(Dane;"???" )	Zwraca liczbę komórek tekstowych zawierających dokładnie trzy znaki.
=LICZ.JEŻELI(Dane;"budżet" )	Zwraca liczbę komórek zawierających wyłącznie pojedyncze słowo budżet, przy czym przy sprawdzaniu nie jest uwzględniana wielkość znaków.
=LICZ.JEŻELI(Dane;"*budżet*")	Zwraca liczbę komórek zawierających słowo budżet w dowolnym miejscu.
=LICZ.JEŻELI(Dane;"A*")	Zwraca liczbę komórek zawierających tekst zaczynający się literą A, przy czym przy sprawdzaniu nie jest uwzględniana wielkość znaków.
=LICZ.JEŻELI(Dane;DZIŚ())	Zwraca liczbę komórek zawierających aktualną datę.
=LICZ.JEŻELI(Dane;">"&ŚREDNIA(Dane))	Zwraca liczbę komórek zawierających wartości większe niż średnia zbioru.
=LICZ.JEŻELI(Dane;">"&ŚREDNIA(Dane)+ ODCH.STANDARDOWE(Dane)*3)	Zwraca liczbę komórek zawierających dane większe od sumy średniej i trzykrotnej wartości odchylenia standardowego.
=LICZ.JEŻELI(Dane;3)+ LICZ.JEŻELI(Dane;-3)	Zwraca liczbę komórek zawierających wartości 3 lub -3.
=LICZ.JEŻELI(Dane;PRAWDA)	Zwraca liczbę komórek zawierających logiczne wartości PRAWDA.
=LICZ.JEŻELI(Dane;PRAWDA)+ LICZ.JEŻELI(Dane;FAŁSZ)	Zwraca liczbę komórek zawierających wartości logiczne (zarówno PRAWDA, jak i FAŁSZ).
=LICZ.JEŻELI(Dane;"#N/D!" )	Zwraca liczbę komórek zawierających wartości błędu #N/D!.



## Sposób 101. Zliczanie komórek spełniających wiele kryteriów jednocześnie

Dzięki lekturze sposobu 100. poznałeś kilkanaście przykładów zastosowania funkcji LICZ.JEŻELI. Formuły te są przydatne w przypadku zliczania komórek spełniających jedno kryterium. Przykłady formuł zaprezentowane tutaj pomogą Ci w zliczeniu komórek, które uwzględniane mają być tylko w przypadkach, gdy spełnione są dwa lub więcej warunków. Kryteria te mogą być tworzone zarówno w oparciu o dane znajdujące się w zliczanych komórkach, jak i informacje pochodzące z innych zakresów komórek.

### Używanie kryteriów połączonych spójnikiem „i”

Zastosowanie iloczynu logicznego kryteriów zliczania spowoduje, że uwzględniane w nim będą tylko te komórki, dla których są spełnione wszystkie określone warunki. Typową sytuacją będzie tu zliczanie wartości mieszczących się w pewnym przedziale liczbowym. Może na przykład zająć konieczność policzenia komórek zawierających dane większe od 0 i mniejsze lub równe wartości 12, co oznacza, że zliczona ma być każda liczba dodatnia mniejsza lub równa 12. Zadanie takie może z powodzeniem wykonać formuła wykorzystująca funkcję LICZ.JEŻELI:

```
=LICZ.JEŻELI(Dane;">0")-LICZ.JEŻELI(Dane;">12")
```

Formuła ta oblicza liczbę wartości większych od zera znajdujących się w określonym zakresie, a następnie odejmuje od niej liczbę danych większych od 12. Wynikiem jest liczba komórek, w których znajdują się dane większe od 0 i mniejsze lub równe 12.

Tworzenie takich formuł może być nieco kłopotliwe, gdyż — jak widać w przytoczonym tu przykładzie — może w nich wystąpić warunek w rodzaju ">12", mimo że celem jest policzenie wartości mniejszych od liczby 12 lub jej równych. Alternatywą może być zastosowanie formuły tablicowej podobnej do zaprezentowanej poniżej. Opracowanie tego typu formuł może Ci się wydawać łatwiejsze:

```
=SUMA((Dane>0)*(Dane<=12))
```



Pamiętaj, że aby wprowadzić formułę tablicową, powinieneś nacisnąć kombinację klawiszy **Ctrl+Shift+Enter** zamiast samego klawisza **Enter**.

Na rysunku 101.1 przedstawiony został prosty arkusz kalkulacyjny, który może zostać zastosowany do prezentacji działania zamieszczonych niżej przykładów formuł. W arkuszu tym zebrano dane dotyczące sprzedaży ułożone według kolejnych miesięcy, dane poszczególnych przedstawicieli handlowych i dane typów klientów. W arkuszu zdefiniowane zostały nazwy odpowiadające nagłówkom kolumn umieszczonym w pierwszym wierszu.

**Rysunek 101.1.**

Arkusz ten stanowi dobry przykład zbioru danych, na którym można przeprowadzać operacje zliczania wartości z wykorzystaniem różnych technik i w oparciu o wiele kryteriów jednocześnie

	A	B	C	D	E
1	<b>Miesiąc</b>	<b>Handlowiec</b>	<b>Typ</b>	<b>Kwota</b>	
2	Styczeń	Albert	Nowy	85	
3	Styczeń	Albert	Nowy	675	
4	Styczeń	Bożyk	Nowy	130	
5	Styczeń	Czaja	Nowy	1350	
6	Styczeń	Czaja	Istniejący	685	
7	Styczeń	Bożyk	Nowy	1350	
8	Styczeń	Czaja	Nowy	475	
9	Styczeń	Bożyk	Nowy	1205	
10	Luty	Bożyk	Istniejący	450	
11	Luty	Albert	Nowy	495	
12	Luty	Czaja	Nowy	210	
13	Luty	Czaja	Istniejący	1050	
14	Luty	Albert	Nowy	140	
15	Luty	Bożyk	Nowy	900	
16	Luty	Bożyk	Nowy	900	
17	Luty	Czaja	Nowy	95	
18	Luty	Czaja	Nowy	780	
19	Marzec	Bożyk	Nowy	900	
20	Marzec	Albert	Istniejący	875	
21	Marzec	Bożyk	Nowy	50	
22	Marzec	Bożyk	Nowy	875	
23	Marzec	Czaja	Istniejący	225	
24	Marzec	Czaja	Nowy	175	
25	Marzec	Bożyk	Istniejący	400	
26	Marzec	Albert	Nowy	840	
27	Marzec	Czaja	Nowy	132	
28					
29					

Czasami kryterium zliczania może być utworzone w oparciu o komórki inne niż te, które podlegają zliczaniu. Możesz na przykład chcieć, aby obliczona została liczba sprzedaży spełniających następujące warunki:

- ♦ *Miesiąc* to Styczeń i
- ♦ *Handlowiec* to Bożyk i
- ♦ *Kwota* jest większa od 1000.

Przedstawiona poniżej formuła tablicowa zwróci liczbę elementów, które spełniają wszystkie trzy podane kryteria:

```
=SUMA((Miesiąc="Styczeń")*(Handlowiec="Bożyk")*(Kwota>1000))
```

## Używanie kryteriów połączonych spójnikiem „lub”

Aby wykorzystać w zliczaniu komórek alternatywę logiczną, wystarczy czasem zastosować wielokrotne wywołanie funkcji LICZ.JEŻELI. Następująca formuła zlicza na przykład wystąpienia wszystkich liczb 1, 3 i 5 wchodzących w skład zakresu *Dane*:

```
=LICZ.JEŻELI(Dane;1)+LICZ.JEŻELI(Dane;3)+LICZ.JEŻELI(Dane;5)
```

Funkcji LICZ.JEŻELI możesz także użyć do utworzenia formuły tablicowej pozwalającej osiągnąć taki sam rezultat:

```
=SUMA(LICZ.JEŻELI(Dane;{1;3;5}))
```

Jeśli jednak spróbujesz skorzystać z alternatywy kryteriów innych niż oparte na wartościach komórek zliczanych, funkcja LICZ.JEŻELI przestanie spełniać swoje zadanie. Powiedzmy, że w zbiorze danych widocznym w arkuszu, który został przedstawiony na rysunku 101.1, będziesz chciał obliczyć liczbę transakcji spełniających następujące kryteria:

- ◆ *Miesiąc* to Styczeń **lub**
- ◆ *Handlowiec* to Bożyk **lub**
- ◆ *Kwota* jest większa od 1000.

Prawidłowy wynik dla takich warunków zwróci przedstawiona niżej formuła tablicowa:

```
=SUMA(JEŻELI((Miesiąc="Styczeń")+(Handlowiec="Bożyk")+(Kwota>1000);1))
```

## Łączenie kryteriów „i” oraz „lub”

W formułach służących do zliczania wystąpień wartości możesz łączyć warunki wykorzystujące alternatywę i iloczyn logiczny. Może Ci się to na przykład przydać do obliczenia liczby transakcji, które spełniają następujące warunki:

- ◆ *Miesiąc* to Styczeń **i**
- ◆ *Handlowiec* to Bożyk **lub** *Handlowiec* to Czaja.

W tym przykładzie dwa warunki dotyczące nazwisk handlowców umieszczone są w jednej linii, aby zaznaczyć, że wszystkie zliczane transakcje muszą być ze stycznia, a ponadto każda musi być wykonana przez jednego z wymienionych handlowców. Następująca formuła tablicowa zwróci liczbę sprzedaży spełniających zadane kryteria:

```
=SUMA((Miesiąc="Styczeń")*JEŻELI((Handlowiec="Bożyk")+(Handlowiec="Czaja");1))
```

## Sposób 102. Obliczanie liczby różnych wpisów w zakresie

Program Excel jest często wykorzystywany do zliczania niepowtarzających się wystąpień danych w pewnym zakresie komórek arkusza.

Najprostszą metodą znalezienia tej liczby jest użycie formuły tablicowej. Podana tu formuła tablicowa zwraca liczbę różnych wpisów znajdujących się w zakresie o nazwie *Dane*:

```
=SUMA(1/LICZ.JEŻELI(Dane;Dane))
```



Aby wprowadzić formułę tablicową, powinieneś nacisnąć kombinację klawiszy *Ctrl+Shift+Enter* zamiast samego klawisza *Enter*.

Na rysunku 102.1 przedstawiono wynik zastosowania formuły tablicowej do obliczenia liczby wystąpień różnych danych w zakresie komórek *A1:C12*. Formuła tablicowa umieszczona w komórce *D3* ma następującą postać:

```
=SUMA(1/LICZ.JEŻELI(A1:C12;A1:C12))
```

Zwraca ona wartość 3, ponieważ w zakresie *A1:C12* występują tylko trzy różne wpisy.

### Rysunek 102.1.

*Umieszczona w komórce D3 formuła tablicowa zlicza wystąpienia różnych wpisów w zakresie komórek*

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F
1	kot	pies	pies			
2	pies	pies	pies			
3	królik	pies	królik	3	zwierzęta	
4	pies	kot	pies			
5	pies	pies	pies			
6	kot	pies	królik			
7	pies	kot	pies			
8	pies	pies	królik			
9	pies	pies	pies			
10	kot	kot	królik			
11	pies	pies	królik			
12	pies	pies	królik			
13						
14						

## Sposób 103. Obliczanie sum warunkowych wykorzystujących pojedynczy warunek

Oferowana przez program Excel funkcja *SUMA* jest jedną z najczęściej używanych funkcji arkusza kalkulacyjnego. Czasami jednak będziesz potrzebował nieco bardziej elastycznych rozwiązań. Z pomocą przyjdzie Ci wówczas funkcja *SUMA.JEŻELI*, która pozwala na tworzenie sum warunkowych. Przyda Ci się ona na przykład w sytuacji, gdy będziesz musiał obliczyć sumę wszystkich liczb ujemnych należących do danego zakresu komórek arkusza.

Zamieszczone tutaj przykłady formuł mają przybliżyć Ci metody korzystania z funkcji *SUMA.JEŻELI* przy opracowywaniu sum warunkowych używających tylko jednego kryterium doboru wartości sumowanych.

Przedstawione tu formuły używają danych znajdujących się w arkuszu kalkulacyjnym, który został pokazany na rysunku 103.1. Arkusz ten zawiera dane dotyczące faktur handlowych. W komórkach w kolumnie *F* umieszczono formuły, których zadaniem jest odejmowanie dat widocznych w kolumnie *E* od dat z komórek kolumny *D*. Ujemne wartości w kolumnie *F* oznaczają, że terminy płatności faktur minęły. W arkuszu zdefiniowano nazwy zakresów, które odpowiadają etykiетom kolumn zamieszczonym w jego pierwszym wierszu (spacje w nazwach kolumn zastąpione zostały w nazwach zakresów znakiem podkreślenia).

**Rysunek 103.1.**

Ujemne wartości w kolumnie *F* oznaczają przekroczenia terminów płatności faktur

	A	B	C	D	E	F	G
1	<b>Nr faktury</b>	<b>Biuro</b>	<b>Kwota</b>	<b>Termin płatności</b>	<b>Dziś</b>	<b>Różnica</b>	
2	AG-0145	Śląskie	5 000,00 zł	04-kwi	07-maj	-33	
3	AG-0189	Lubuskie	450,00 zł	22-kwi	07-maj	-15	
4	AG-0220	Mazowieckie	3 211,56 zł	01-maj	07-maj	-6	
5	AG-0310	Śląskie	250,00 zł	03-maj	07-maj	-4	
6	AG-0355	Mazowieckie	125,50 zł	07-maj	07-maj	0	
7	AG-0409	Mazowieckie	3 000,00 zł	13-maj	07-maj	6	
8	AG-0581	Śląskie	2 100,00 zł	26-maj	07-maj	19	
9	AG-0600	Śląskie	335,39 zł	26-maj	07-maj	19	
10	AG-0602	Mazowieckie	65,00 zł	31-maj	07-maj	24	
11	AG-0633	Lubuskie	250,00 zł	02-cze	07-maj	26	
12	SUMA		14 787,45 zł			36	
13							
14							

## Sumowanie wyłącznie wartości ujemnych

Podana niżej formuła zwraca sumę wszystkich wartości ujemnych znajdujących się w kolumnie *F*. Innymi słowy, zwraca ona sumaryczną liczbę dni opóźnienia w płatnościach wszystkich faktur. W przypadku przedstawionego tu przykładowego arkusza wartość ta wyniesie -58.

```
=SUMA.JEŻELI(Różnica;"<0")
```

Funkcja SUMA.JEŻELI może przyjąć trzy argumenty. Ponieważ nie określasz tu trzeciego argumentu, drugi parametr (" $<0$ ") zostanie zastosowany w odniesieniu do zakresu o nazwie *Różnica*.

## Sumowanie wartości w oparciu o inny zakres

Następująca formuła zwraca sumę kwot (pobranej z kolumny *C*) wszystkich faktur, których terminy płatności zostały przekroczone:

```
=SUMA.JEŻELI(Różnica;"<0";Kwota)
```

Formuła ta korzysta z wartości znajdujących się w zakresie *Różnica* do określenia, które z danych należących do zakresu *Kwota* powinny zostać zsumowane.

## Sumowanie wartości w oparciu o porównanie tekstowe

Zamieszczona poniżej formuła zwraca sumę kwot wszystkich faktur wystawionych przez śląskie biuro firmy:

```
=SUMA.JEŻELI(Biuro;"=Śląskie";Kwota)
```

Użycie drugiego znaku równości jest opcjonalne. Następująca formuła zwróci dokładnie ten sam wynik:

```
=SUMA.JEŻELI(Biuro;"Śląskie";Kwota)
```

Jeśli będziesz chciał zsumować kwoty faktur wystawionych przez wszystkie biura przedsiębiorstwa **oprócz** biura śląskiego, powinieneś skorzystać z formuły:

```
=SUMA.JEŻELI(Biuro:"<Śląskie";Kwota)
```

## Sumowanie wartości w oparciu o porównanie dat

Przedstawiona niżej formuła zwraca sumaryczną wartość kwot wszystkich faktur, których terminy płatności przypadają na datę występującą po 1 czerwca 2005 roku:

```
=SUMA.JEŻELI(Termin_płatności;">="&DATA(2005;6;1);Kwota)
```

Zwróć uwagę na fakt, że drugim argumentem funkcji SUMA.JEŻELI jest wyrażenie. Wyrażenie to korzysta z funkcji DATA, która zwraca datę, ta zaś jest połączona z operatorem porównania (umieszczonym w cudzysłowach) za pomocą operatora konkatencji (&).

Podana poniżej formuła zwraca sumę kwot faktur, które mają przyszłą datę płatności, włączając w to dzień dzisiejszy:

```
=SUMA.JEŻELI(Termin_płatności;">="&DZIŚ());Kwota)
```

## Sposób 104. Obliczanie sum warunkowych wykorzystujących wiele warunków

Powyżej przedstawiono szereg przykładów sumowania warunkowego używającego tylko jednego warunku do sprawdzania wartości podlegających dodawaniu. Przykłady zamieszczone tutaj dotyczą sumowania warunkowego opartego na wielu kryteriach. Funkcja SUMA.JEŻELI nie pozwala na definiowanie większej liczby warunków, dlatego w takich sytuacjach będziesz musiał skorzystać z formuł tablicowych.

Na rysunku 104.1 przedstawiony został znany Ci z poprzedniego sposobu przykładowy arkusz zawierający dane o fakturach. Nie oznacza to jednak, oczywiście, że zamieszczonych tu formuł nie będziesz mógł zastosować w swoich arkuszach i dopasować do własnych potrzeb.

## Używanie kryteriów połączonych spójnikiem „i”

Załóżmy, że chcesz otrzymać sumę kwot faktur, które nie zostały zapłacone w terminie **i** były wystawione przez biuro śląskie. Innymi słowy, chodzi Ci o to, aby dana pochodząca z zakresu *Kwota* została uwzględniona podczas dodawania tylko wtedy, gdy obydwa z wymienionych niżej kryteriów będą spełnione:

- ♦ odpowiadająca jej liczba z zakresu *Różnica* ma wartość ujemną,
- ♦ odpowiadający jej tekst z zakresu *Biuro* ma postać: Śląskie.

**Rysunek 104.1.**

Na przykładzie przedstawionych tu danych zostanie działanie sumowania warunkowego korzystającego z wielu kryteriów

	A	B	C	D	E	F	G
1	<b>Nr faktury</b>	<b>Biuro</b>	<b>Kwota</b>	<b>Termin płatności</b>	<b>Dziś</b>	<b>Różnica</b>	
2	AG-0145	Śląskie	5 000,00 zł	04-kwi	07-maj	-33	
3	AG-0189	Lubuskie	450,00 zł	22-kwi	07-maj	-15	
4	AG-0220	Mazowieckie	3 211,56 zł	01-maj	07-maj	-6	
5	AG-0310	Śląskie	250,00 zł	03-maj	07-maj	-4	
6	AG-0355	Mazowieckie	125,50 zł	07-maj	07-maj	0	
7	AG-0409	Mazowieckie	3 000,00 zł	13-maj	07-maj	6	
8	AG-0581	Śląskie	2 100,00 zł	26-maj	07-maj	19	
9	AG-0600	Śląskie	335,39 zł	26-maj	07-maj	19	
10	AG-0602	Mazowieckie	65,00 zł	31-maj	07-maj	24	
11	AG-0633	Lubuskie	250,00 zł	02-cze	07-maj	26	
12	SUMA		14 787,45 zł			36	
13							
14							

Określone w ten sposób zadanie wykona następująca formuła tablicowa:

```
=SUMA((Różnica<0)*(Biuro="Śląskie")*Kwota)
```



Formułę tablicową wprowadza się przy użyciu kombinacji klawiszy *Ctrl+Shift+Enter*.

Nietablicową alternatywą dla tej formuły może być następująca formuła:

```
=SUMA.ILOCZYŃÓW(1*(Różnica<0);1*(Biuro="Śląskie");Kwota)
```

## Używanie kryteriów połączonych spójnikiem „lub”

Wyobraź sobie, że Twoim zadaniem jest obliczenie sumy kwot takich faktur, które nie zostały zapłacone w terminie **lub** są związane ze śląskim biurem firmy. Inaczej mówiąc, wartości należące do zakresu *Kwota* zostaną wykorzystane do tworzenia sumy w sytuacji, gdy spełniony jest choć jeden z warunków:

- ♦ odpowiadająca im liczba z zakresu *Różnica* ma wartość ujemną,
- ♦ odpowiadający im tekst z zakresu *Biuro* ma postać: *Śląskie*.

Określone w ten sposób zadanie wykona następująca formuła tablicowa:

```
=SUMA(JEŻELI((Biuro="Śląskie")+ (Różnica<0);1;0)*Kwota)
```

Znak dodawania (+) łączy obydwie warunki i jeśli chcesz uwzględnić więcej kryteriów, powinieneś po prostu dodać kolejne warunki za jego pomocą.

## Używanie kryteriów połączonych spójnikami „i” oraz „lub”

Sprawy nieco się komplikują, gdy zachodzi potrzeba połączenia kryteriów zarówno za pomocą alternatywy, jak i iloczynu logicznego. Może na przykład zająć konieczność zsumowania takich wartości pochodzących z zakresu *Kwota*, dla których spełnione są oba wymienione niżej warunki:

- ♦ odpowiadająca im liczba z zakresu *Różnica* ma wartość ujemną,
- ♦ odpowiadający im tekst z zakresu *Biuro* ma postać *Śląskie* lub *Lubuskie*.

Zauważ, że drugie z kryteriów składa się tak naprawdę z dwóch warunków połączonych spójnikiem „lub”. Rozwiązaniem będzie tu następująca formuła tablicowa:

```
=SUMA((Różnica<0)*JEŻELI((Biuro="Śląskie")+ (Biuro="Lubuskie");1)*Kwota)
```

## Sposób 105. Wyszukiwanie wartości dokładnej

Funkcje Excela WYSZUKAJ.PIONOWO i WYSZUKAJ.POZIOMO są bardzo przydatne w sytuacjach, gdy musisz pobrać daną ze znajdującej się w zakresie komórek tabeli, wyszukując pewną inną wartość.

Klasyczny przykład wykorzystania funkcji wyszukiwania przedstawiony na rysunku 105.1 stanowi sprawdzanie stopy podatkowej stosowanej przy określonej kwocie dochodów. Tabela wysokości stóp podatkowych zawiera wartości, które odpowiadają pewnym przedziałom rocznych zarobków. Następująca formuła umieszczona w komórce B3 przedstawionego arkusza pozwala określić, jaką stopę należy zastosować dla wartości dochodu wpisanej do komórki B2:

```
=WYSZUKAJ.PIONOWO(B2;D2:F7;3)
```

**Rysunek 105.1.**

*Korzystanie z funkcji WYSZUKAJ.PIONOWO do odnalezienia odpowiedniej stopy podatkowej*

	A	B	C	D	E	F	G
				Dochód jest równy lub większy niż...	Leczniejszy niż...	Stopa podatkowa	
1							
2		Wprowadź dochód:	64 500 zł	0 zł	2 650 zł	15,00%	
3		Stopa podatkowa wynosi:	36,00%	2 651 zł	27 300 zł	28,00%	
4				27 301 zł	58 500 zł	31,00%	
5				58 501 zł	131 800 zł	36,00%	
6				131 801 zł	284 700 zł	39,60%	
7				284 701 zł		45,25%	
8							
9							
10							

Przytoczony tu przykład pokazuje, że funkcje WYSZUKAJ.PIONOWO i WYSZUKAJ.POZIOMO nie wymagają znalezienia dokładnej wartości wśród danych przeszukiwanego zbioru. Jeśli funkcje nie znajdą dokładnej wartości, zwrócą dane związane z największą wartością, ale mniejszą od poszukiwanej (dane w kolumnie D powinny być posortowane w porządku rosnącym). W niektórych przypadkach to Ty będziesz **wymagał** odnalezienia wartości spełniającej dokładnie zadane kryterium wyszukiwania. Będzie tak na przykład w sytuacji, gdy będziesz szukał określonego numeru pracownika.

Aby odnaleźć dane dokładnie spełniające podane kryterium, skorzystaj z dodatkowego czwartego argumentu wywołania funkcji WYSZUKAJ.PIONOWO lub WYSZUKAJ.POZIOMO. Jest to argument opcjonalny i powinien mieć wówczas wartość FAŁSZ.



Na rysunku 105.2 przedstawiony został arkusz zawierający tabelę, w której umieszczono są numery pracowników (w kolumnie *C*) i ich nazwiska (w kolumnie *D*). Tabela ta nosi nazwę *ListaPracowników*. Formuła, która znajduje się w komórce *B2*, przeszukuje tabelę w celu znalezienia numeru pracownika podanego w komórce *B1* i zwraca odpowiednie dla niego nazwisko. Ma ona postać:

=WYSZUKAJ.PIONOWO(B1;ListaPracowników;2;FAŁSZ)

### Rysunek 105.2.

*Wyszukiwanie w przedstawionej tabeli wymaga zastosowania dokładnych porównań danych*

	A	B	C	D
1	<b>Nr pracownika:</b>	900	<b>Nr pracownika</b>	<b>Nazwisko pracownika</b>
2	<b>Nazwisko pracownika:</b>	#N/D!	345	Anatol Tolski
3			1021	Anatol Tolski
4			405	Czesław S. Biłski
5			1093	Czesław S. Biłski
6			412	Czesława Brzdąc
7			543	Dariusz Mariusz
8			590	Dobroslawa Wolfska
9			602	Jan Chrzan
10			593	Janina Jedliga
11			661	Janusz Bębenek
12			732	Józef Skrzep
13			784	Kamil Tyteczek
14			801	Krystyna Kamiś
15			822	Michał Hętko
16			854	Przemysław Dobczyk
17			875	Rajmund Chłopek
18			901	Robert H. Miły
19			932	Roman Strup
20			987	Stanisław Zdzisław
21			438	Stefan H. Kac
22			871	Tobiasz E. Absynt
23			1032	Tomasz Błąd
24				

Z uwagi na to, że ostatni argument funkcji WYSZUKAJ.PIONOWO ma logiczną wartość FAŁSZ, zwraca ona nazwisko pracownika tylko wtedy, gdy znajdzie jego numer w pełni odpowiadający wartości podanej jako kryterium. W innym przypadku formuła zwraca kod błędu #N/D!. Jest to, oczywiście, działanie jak najbardziej pożądane, ponieważ otrzymanie wyniku przybliżonego przy wyszukiwaniu określonego numeru pracownika zupełnie mija się z celem. Zwróć również uwagę na fakt, że numery pracowników znajdujące się w kolumnie *C* nie są ułożone w kolejności rosnącej. Jeżeli ostatni argument funkcji WYSZUKAJ.PIONOWO ma wartość FAŁSZ, przeszukiwane wartości nie muszą być ułożone w porządku rosnącym.

Jeśli w sytuacji, gdy nie zostanie znaleziony poszukiwany numer pracownika, w komórce wyniku wolisz oglądać coś innego niż kod błędu #N/D!, powinieneś skorzystać z funkcji CZY.BRAK w celu sprawdzenia, czy rezultatem działania funkcji WYSZUKAJ.PIONOWO nie jest ten właśnie błąd, a następnie użyć funkcji JEŻELI do zastąpienia go jakimś innym tekstem. Zamieszczona poniżej formuła wykonuje to zadanie, zastępując kod błędu #N/D! informacją Nie znaleziono:

=JEŻELI(CZY.BRAK(WYSZUKAJ.PIONOWO(B1;ListaPracowników;2;FAŁSZ)));  
"Nie znaleziono";WYSZUKAJ.PIONOWO(B1;ListaPracowników;2;FAŁSZ))

## Sposób 106. Przeprowadzanie wyszukiwań dwuwymiarowych

Wyszukiwanie dwuwymiarowe polega na odnajdowaniu wartości na przecięciu pewnej kolumny i wiersza. Tutaj opisano dwie metody przeprowadzania tego typu wyszukiwań.

### Użycie formuły

Na rysunku 106.1 przedstawiono arkusz kalkulacyjny, w którym znajduje się tabela zawierająca wartości sprzedaży osiągnięte w poszczególnych miesiącach dla różnych kategorii produktów. Aby poznać dane dotyczące sprzedaży określonego towaru w wybranym miesiącu, użytkownik powinien wpisać nazwę miesiąca do komórki *B1*, a nazwę produktu do komórki *B2*.

**Rysunek 106.1.**

Tabela przedstawiająca zasadę działania wyszukiwania dwuwymiarowego

	A	B	C	D	E	F	G	H
1	Miesiąc:	Lipiec		Zabawki	Przytulanki	Wózki	Zsumowane	
2	Produkt:	Przytulanki	Styczeń	3 221	1 531	3 198	7 950	
3			Luty	4 314	4 510	4 322	13 146	
4	Pozycja miesiąca:	8	Marzec	3 212	1 231	3 322	7 765	
5	Pozycja produktu:	3	Kwiecień	5 342	2 185	1 981	9 508	
6	Wartość sprzedaży:	2 185	Maj	1 231	3 198	5 342	9 771	
7			Czerwiec	3 122	1 172	1 231	5 525	
8			Lipiec	3 198	2 185	3 122	8 505	
9			Sierpień	1 172	1 981	1 172	4 325	
10			Wrzesień	2 185	3 122	2 185	7 492	
11			Październik	2 156	1 521	1 242	4 919	
12			Listopad	1 114	2 211	2 163	5 488	
13			Grudzień	3 981	5 912	2 141	12 034	
14			Suma	34 248	30 759	31 421	96 428	
15								

W celu uproszczenia działań w arkuszu zdefiniowano następujące nazwy:

Nazwa	Odnosi się do
<i>Miesiąc</i>	<i>B1</i>
<i>Produkt</i>	<i>B2</i>
<i>Tabela</i>	<i>D1:H14</i>
<i>ListaMiesiący</i>	<i>D1:D14</i>
<i>ListaProduktów</i>	<i>D1:H1</i>

Podana niżej formuła (umieszczona w arkuszu w komórce *B4*) korzysta z funkcji `PODAJ.POZYCJĘ` do pobrania pozycji miesiąca w zakresie *ListaMiesiący*. Jeśli więc na przykład szukany miesiąc będzie Styczeń, formuła zwróci liczbę 2, gdyż Styczeń jest drugim elementem wchodzącym w skład zakresu *ListaMiesiący*. Jego pierwszym elementem jest bowiem pusta komórka *D1*.

`=PODAJ.POZYCJĘ(Miesiąc;ListaMiesiący;0)`

Formuła znajdująca się w komórce *B5* działa w podobny sposób, różnica polega tu tylko na tym, że przeszukuje ona zakres *ListaProduktów* w celu sprawdzenia pozycji określonej kategorii towaru.

```
=PODAJ.POZYCJĘ(Produkt;ListaProduktów;0)
```

Ostateczna formuła umieszczona w komórce *B6* zwraca odpowiednią wartość sprzedaży. Wykorzystuje w tym celu funkcję *INDEKS*, podając zawartość komórek *B4* i *B5* w charakterze jej argumentów.

```
=INDEKS(Tabela;B4;B5)
```

Możesz, oczywiście, połączyć wszystkie trzy wymienione wyżej formuły w jedną, otrzymując następującą formułę, która zwróci, rzecz jasna, ten sam wynik:


```
=INDEKS(Tabela;PODAJ.POZYCJĘ(Miesiąc;ListaMiesiący;0);PODAJ.POZYCJĘ(Produkt;ListaProduktów;0))
```



Formuły tego typu możesz również tworzyć za pomocą przedstawionego na rysunku 106.2 narzędzia *Kreator odnośników*, będącego jednym ze standardowych i rozpowszechnianych wraz z aplikacją dodatków do programu Excel. Aby zainstalować ten dodatek, powinieneś wybrać z menu polecenie *Narzędzia/Dodatki...* Po zainstalowaniu narzędzia będziesz mógł uruchomić je, wybierając z menu polecenie *Narzędzia/Kreator/Odnośników...*

### Rysunek 106.2.

*Dodatek Kreator odnośników umożliwia utworzenie formuły służącej do przeprowadzania wyszukiwania dwuwymiarowego*



## Użycie bezpośredniego przecięcia

Druga metoda przeprowadzania wyszukiwania dwuwymiarowego jest dużo prostsza, ale wymaga wcześniejszego utworzenia nazw dla każdego wiersza i dla każdej kolumny tabeli danych.

Sposobem na szybkie nazwanie wszystkich wierszy i kolumn jest zaznaczenie całej tabeli i wybranie z menu polecenia *Wstaw/Nazwa/Utwórz...*, a następnie zaznaczenie odpowiednich opcji w oknie *Tworzenie nazw*. Po utworzeniu nazw powinieneś skorzystać z prostej formuły, która wykona dla Ciebie wyszukiwanie dwuwymiarowe i będzie miała następującą postać:

=Przytulanki Lipiec


Formuła ta używa operatora przecięcia zakresów, którym jest spacja. Jej wpisanie spowoduje w tym przypadku zwrócenie osiągniętej w lipcu wartości sprzedaży artykułów należących do kategorii Przytulanki.

## Sposób 107. Przeprowadzanie wyszukiwania w dwóch kolumnach

Niektóre sytuacje wymagają wyszukiwania prowadzonego jednocześnie w dwóch kolumnach wartości. Na rysunku 107.1 przedstawiono przykład arkusza, w którym istnieje konieczność przeprowadzenia takiego właśnie wyszukiwania.

### Rysunek 107.1.

Formuła umieszczona w tym arkuszu prowadzi wyszukiwanie w oparciu o wartości znajdujące się w dwóch kolumnach (D i E)



	A	B	C	D	E	F	G
1	<b>Marka:</b>	Jeep		<b>Marka</b>	<b>Model</b>	<b>Kod</b>	
2	<b>Model:</b>	Grand Cherokee		Chevy	Blazer	C-094	
3	<b>Kod:</b>	J-701		Chevy	Tahoe	C-823	
4				Ford	Explorer	F-772	
5				Ford	Expedition	F-229	
6				Isuzu	Rodeo	I-897	
7				Isuzu	Tropper	I-900	
8				Jeep	Cherokee	J-983	
9				Jeep	Grand Cherokee	J-701	
10				Nissan	Pathfinder	N-231	
11				Toyota	4Runner	T-871	
12				Toyota	Land Cruiser	T-981	
13							

Widoczna w arkuszu tabela zawiera kolumny, z których jedna przechowuje informacje o producentach samochodów, druga o modelach pojazdów, w trzeciej zaś umieszczono odpowiednie dla nich kody. Opisana tu technika pozwoli Ci wyszukiwać wartości kodów w oparciu zarówno o markę, jak i model samochodu.

W arkuszu zdefiniowano następujące zakresy nazwane:

Nazwa	Odnosi się do
Marka	B1
Model	B2
Kod	F2:F12
Zakres1	D2:D12
Zakres2	E2:E12

Znajdująca się poniżej formuła tablicowa pozwala na znalezienie kodu odpowiadającego podanej marce i modelowi samochodu:

=INDEKS(Kod;PODAJ.POZYCJĘ(Marka&Model;Zakres1&Zakres2;0))



Pamiętaj, że aby wprowadzić formułę tablicową, powinieneś nacisnąć kombinację klawiszy *Ctrl+Shift+Enter* zamiast samego klawisza *Enter*.

Działanie tej formuły opiera się na połączeniu zawartości komórek *Marka* i *Model* i wyszukiwaniu otrzymanego w ten sposób tekstu w tablicy tekstów utworzonych z połączonych w podobny sposób danych pochodzących z zakresów *Zakres1* i *Zakres2*.

## Sposób 108. Przeprowadzanie wyszukiwania przy użyciu tablicy

Jeśli przeszukiwana przez Ciebie tabela danych ma niewielkie rozmiary, możesz uniknąć stosowania specjalnej tabeli wyszukiwania i wszystkie potrzebne podczas tej czynności informacje przechowywać w tablicy. Opisano tu typowy problem wyszukiwania, który został najpierw rozwiązany za pomocą standardowej tabeli wyszukiwania, a następnie przy użyciu alternatywnej wobec niej tablicy.

### Użycie tabeli wyszukiwania

Na rysunku 108.1 przedstawiono arkusz kalkulacyjny zawierający wyniki testów uczniów pewnej klasy. Zakres komórek *E2:F6*, noszący nazwę *ListaOcen*, stanowi tabelę wyszukiwania. Jest ona używana do przypisania wynikom sprawdzianu odpowiednich ocen.

**Rysunek 108.1.**

*Przypisywanie odpowiednich ocen do wyników sprawdzianu*

	A	B	C	D	E	F	G
1	<b>Uczeń</b>	<b>Wynik</b>	<b>Ocena</b>		<b>Wynik</b>	<b>Ocena</b>	
2	Adamski	36	F		0	F	
3	Baczyński	68	D		40	D	
4	Czachowski	50	D		70	C	
5	Dąbek	77	C		80	B	
6	Gądek	92	A		90	A	
7	Hermanowski	100	A				
8	Jankowski	74	C				
9	Maślak	45	D				
10	Panda	60	D				
11	Rakowski	89	B				
12	Socha	99	A				
13	Tomeczek	91	A				
14	Waleński	59	D				
15							

W komórkach kolumny *C* umieszczono formuły korzystające z funkcji *WYSZUKAJ.PIONOWO* i tabeli wyszukiwania, za pomocą której wynikom znajdującym się w kolumnie *B* przypisywane są właściwe oceny. Formuła przechowywana w komórce *C2* ma na przykład postać:

```
=WYSZUKAJ.PIONOWO(B2;ListaOcen;2)
```

## Użycie tablicy

W sytuacjach, gdy tabela wyszukiwania jest niewielka (tak jak w niniejszym przykładzie), zamiast z niej możesz skorzystać z tablicy dosłownej. Pozwoli to usunąć nadmiar informacji z Twojego arkusza, a korzystająca z tego rozwiązania formuła zwróci wynik bez odwoływania się do tabeli wyszukiwania. Zamiast tego tabela ta będzie niejako zakodowana na stałe w ciele formuły w postaci tablicy stałych wartości. Zwróć uwagę na format zapisu tablicy oraz znaki używane w jej definicji. Tablica oznaczana jest nawiasami klamrowymi ({ i }), poszczególne elementy wierszy oddzielane są za pomocą średników (;), zaś kolejne wiersze oddziela znak odwróconego ukośnika (\).

```
=WYSZUKAJ.PIONOWO(B2;{0;"F"\40;"D"\70;"C"\80;"B"\90;"A"};2)
```

Nieco inną metodą poradzenia sobie z tym zadaniem jest wykorzystanie bardziej czytelnej formuły, w której użyta została funkcja WYSZUKAJ oraz dwa argumenty tablicowe:

```
=WYSZUKAJ(B2;{0;40;70;80;90};{"F";"D";"C";"B";"A"})
```

## Sposób 109. Używanie funkcji ADR.POŚR

Aby uczynić swoje formuły bardziej uniwersalnymi, możesz skorzystać z oferowanej przez program Excel funkcji ADR.POŚR. Umożliwia ona tworzenie odwołań do zakresów komórek. Ta rzadko używana funkcja stosowana jest do zamieniania argumentu tekstowego opisującego odwołanie do pewnego obszaru arkusza na normalne odwołanie do zakresu komórek. Zrozumienie działania funkcji ADR.POŚR bez wątpienia pozwoli Ci na tworzenie bardziej zaawansowanych i interaktywnych arkuszy kalkulacyjnych.

Na rysunku 109.1 pokazano prosty przykład arkusza kalkulacyjnego, w którego komórce E5 umieszczona została następująca formuła:

```
=SUMA(ADR.POŚR("B"&E2&" :B"&E3))
```

### Rysunek 109.1.

*Użycie funkcji ADR.POŚR do zsumowania wartości pochodzących z wierszy podanych przez użytkownika*

	A	B	C	D	E
1	Miesiąc	Wartość sprzedaży			
2	Sty	45 322		Pierwszy wiersz	2
3	Lut	38 900		Ostatni wiersz	4
4	Mar	59 832			
5	Kwi	65 985		<b>SUMA:</b>	<b>144 054</b>
6	Maj	45 001			
7	Cze	78 323			
8	Lip	69 212			
9	Sie	72 090			
10	Wrz	69 832			
11	Paź	69 874			
12	Lis	71 293			
13	Gru	43 505			
14					

Zwróć uwagę na fakt, że w argumencie funkcji ADR.POŚR skorzystano z operatora konkatencji do utworzenia odwołania do zakresu komórek za pomocą wartości umieszczonych w komórkach E2 i E3. Jeśli zatem do pierwszej z nich wprowadzisz liczbę 2, zaś do drugiej wartość 4, argument ten przyjmie postać następującego łańcucha tekstowego:

```
"B2:B4"
```

Funkcja konwertująca zmieni ten ciąg znaków na zwykłe odwołanie do zakresu komórek, które zostanie następnie przekazane do funkcji SUMA w charakterze argumentu. Formuła zwróci zatem taką samą wartość, jak formuła:


```
=SUMA(B2:B4)
```

Wprowadzenie jakichkolwiek zmian do komórek E2 i E3 spowoduje aktualizację odwołania i zmianę formuły, która obliczać będzie zawsze sumę wartości z określonych przez te komórki wierszy kolumny B.

Na rysunku 109.2 przedstawiono kolejny przykład, w którym zastosowano pełne odwołanie wraz z częścią określającą arkusz.

### Rysunek 109.2.

*Wykorzystanie funkcji ADR.POŚR do tworzenia odwołań do zakresów znajdujących się w innych arkuszach skoroszytu*



	A	B	C	D	E	F	G
1	Region	Suma					
2	Północ	473					
3	Południe	631					
4	Zachód	451					
5	Wschód	753					
6							
7							
8							
9							
10							

W kolumnie A arkusza *Podsumowanie* znajdują się wartości tekstowe odpowiadające pozostałym arkuszom wchodzącym w skład bieżącego skoroszytu. W komórkach kolumny B z kolei umieszczone zostały formuły, które odwołują się do tych pozycji. Formuła w komórce B2 ma na przykład postać:

```
=SUMA(ADR.POŚR(A2&"!F1:F10"))
```

Argument funkcji ADR.POŚR powstaje w wyniku połączenia tekstu umieszczonego w komórce A2 z odwołaniem do zakresu podanym w cudzysłowie. Jest on następnie zamieniany przez tę funkcję na odwołanie do zakresu komórek, który wykorzystywany jest z kolei jako argument funkcji SUMA. Formuła jest w tym momencie równoznaczna z następującą:

```
=SUMA(Północ!F1:F10)
```

Formuła ta została skopiowana do kolejnych komórek kolumny arkusza. Każda z widocznych formuł zwraca sumę wartości umieszczonych w zakresie F1:F10 odpowiednich arkuszy kalkulacyjnych.

## Sposób 110. Tworzenie megaformuł

Tutaj znajdziesz opis metody łączenia kilku pośrednich formuł w celu otrzymania jednej długiej formuły, czyli tak zwanej **megaformuły**. W przeszłości z pewnością widziałeś bardzo długie formuły, które były po prostu niemożliwe do zrozumienia. Teraz nauczysz się sam je tworzyć.

Celem jest tutaj opracowanie pojedynczej formuły, której działanie ma polegać na usuwaniu drugich imion z wpisów zawierających imiona i nazwiska. Będzie więc ona na przykład przetwarzała nazwisko Marian Antoni Królik do postaci Marian Królik. Na rysunku 110.1 przedstawiono arkusz kalkulacyjny zawierający zbiór nazwisk oraz sześć kolumn komórek przechowujących formuły pośrednie, których połączenie daje w wyniku zamierzony efekt. Zauważ, że formuły nie są doskonałe i nie radzą sobie na przykład z nazwiskami składającymi się z jednego tylko wyrazu.

**Rysunek 110.1.** Proces usuwania drugich imion lub ich inicjałów wymaga zastosowania sześciu formuł pośrednich

Formuły zebrane zostały w znajdującej się poniżej tabeli 110.1.

**Tabela 110.1.** Formuły pośrednie

Komórka	Formuła pośrednia	Wykonywane działanie
<i>B1</i>	=USUŃ. ZBĘDNE. ODSTĘPY(A1)	Usuwa nadmiarowe znaki odstępu.
<i>C1</i>	=ZNAJDŹ(" ";B1;1)	Znajduje pierwszy znak spacji.
<i>D1</i>	=ZNAJDŹ(" ";B1;C1+1)	Znajduje drugi znak spacji, jeśli taki występuje.
<i>E1</i>	=JEŻELI(CZY.BŁĄD(D1):C1;D1)	Używa pierwszej spacji, jeśli druga nie istnieje.
<i>F1</i>	=LEWY(B1;C1-1)	Wyodrębnia imię.
<i>G1</i>	=PRAWY(B1;DŁ(B1)-E1)	Wyodrębnia nazwisko.
<i>H1</i>	=F1&" "&G1	Łączy imię i nazwisko.

Poświęcając nieco pracy, możesz wyeliminować wszystkie te pośrednie formuły i zastąpić je jedną megaformułą. Cel ten osiągniesz, tworząc najpierw formuły pośrednie, a następnie edytując końcową formułę (w tym przypadku znajdującą się w kolumnie *H*), w której każde odwołanie do komórek formuł powinieneś zamienić na bezpośrednie wywołanie odpowiedniej formuły. Na szczęście masz możliwość skorzystania ze schowka do kopiowania i wklejania formuł, w innym razie musiałbyś bowiem wszystkie przepisać ręcznie. Powtarzaj tę operację do momentu, aż w komórce *H1* nie znaj-



dzie się żadne odwołanie oprócz odwołań do komórki *A1* przechowującej daną wejściową. W wyniku tego działania powinieneś otrzymać następującą megaformułę:

```
=LEWY(USUŃ.ZBĘDNE.ODSTĘPY(A1);ZNAJDŹ(" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);1)-1)&
" "&PRAWY(USUŃ.ZBĘDNE.ODSTĘPY(A1);DŁ(USUŃ.ZBĘDNE.ODSTĘPY(A1))-
JEŻELI(CZY.BŁĄD(ZNAJDŹ(" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);ZNAJDŹ(
" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);1)+1));ZNAJDŹ(" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);1);ZNAJDŹ(
" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);ZNAJDŹ(" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);1)+1)))
```

Gdy będziesz już zadowolony z efektu działania tej megaformuły, będziesz mógł swobodnie usunąć wszystkie kolumny przechowujące formuły pośrednie, ponieważ nie będą więcej do niczego potrzebne. Jeśli wciąż nie jesteś pewien, czy dobrze zrozumiałeś przedstawioną tu procedurę, przeczytaj uważnie kolejne kroki, które trzeba wykonać, aby otrzymać megaformułę:

1. Spójrz na zawartość komórki *H1*. Znajdują się w niej dwa odwołania do komórek *F1* i *G1*:

```
=F1&" "&G1
```

2. Przejdź do komórki *G1* i skopiuj umieszczoną w niej treść formuły do schowka, pomijając znak równości.
3. Wróć do komórki *H1* i zastąp widoczne w niej odwołanie do komórki *G1* formułą skopiowaną przed chwilą do schowka. W komórce *H1* powinna w tym momencie znajdować się następująca formuła:

```
=F1&" "&PRAWY(B1;DŁ(B1)-E1)
```

4. Przejdź do komórki *F1* i skopiuj umieszczoną w niej treść formuły do schowka, pomijając znak równości.
5. Wróć do komórki *H1* i zastąp występujące w niej odwołanie do komórki *F1* formułą skopiowaną przed chwilą do schowka. W komórce *H1* powinna w tym momencie znajdować się następująca formuła:

```
=LEWY(B1;C1-1)&" "&PRAWY(B1;DŁ(B1)-E1)
```

6. W komórce *H1* występują w tym momencie odwołania do trzech komórek arkusza, a mianowicie do komórek *B1*, *C1* i *E1*. Odwołania te powinieneś zastąpić, używając formuł znajdujących się w odpowiednich komórkach.

7. Zastąp odwołanie do komórki *E1* przechowywaną w niej formułą. Wynik tego działania powinien być następujący:

```
=LEWY(B1;C1-1)&" "&PRAWY(B1;DŁ(B1)-JEŻELI(CZY.BŁĄD(D1);C1;D1))
```

8. Zwróć uwagę, że formuła znajdująca się obecnie w komórce *H1* zawiera dwa odwołania do komórki *D1*. Skopiuj formułę umieszczoną w tej komórce i zastąp nią obydwie odwołania. Formuła przyjmie po tym następującą postać:

```
=LEWY(B1;C1-1)&" "&PRAWY(B1;DŁ(B1)-JEŻELI(CZY.BŁĄD(ZNAJDŹ(
" ";B1;C1+1));C1;ZNAJDŹ(" ";B1;C1+1)))
```

9. Zastąp wszystkie cztery odwołania do komórki *C1* znajdującą się w niej formułą. Formuła w komórce *H1* przyjmie wówczas postać:

```
=LEWY(B1;ZNAJDŹ(" ";B1;1)-1)&" "&PRAWY(B1;DŁ(B1)-JEŻELI(CZY.BŁĄD(ZNAJDŹ(
" ";B1;ZNAJDŹ(" ";B1;1)+1));ZNAJDŹ(" ";B1;1);ZNAJDŹ(" ";B1;ZNAJDŹ(
" ";B1;1)+1)))
```

- 10.** Ostatnim krokiem będzie zastąpienie dziewięciu wystąpień odwołania do komórki *B1* za pomocą przechowywanej w niej formuły. W wyniku tego działania powinieneś otrzymać szukaną megaformułę:

```
=LEWY(USUŃ.ZBĘDNE.ODSTĘPY(A1);ZNAJDŹ(" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);1)-1)&
" "&PRAWY(USUŃ.ZBĘDNE.ODSTĘPY(A1);DŁ(USUŃ.ZBĘDNE.ODSTĘPY(A1))-
JEŻELI(CZY.BŁĄD(ZNAJDŹ(" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);ZNAJDŹ(
" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);1)+1));ZNAJDŹ(
" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);1);ZNAJDŹ(" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);ZNAJDŹ(
" ";USUŃ.ZBĘDNE.ODSTĘPY(A1);1)+1)))
```

Zauważ, że ostateczna wersja formuły umieszczona w komórce *H1* zawiera odwołania tylko i wyłącznie do komórki *A1*. Tworzenie megaformuły zostało zatem ukończone, a jej działanie będzie dokładnie odpowiadało zestawowi czynności wykonywanych przez formuły pośrednie, które możesz teraz spokojnie usunąć z arkusza.

Technikę tę możesz, oczywiście, zastosować w przypadku opracowywania swoich własnych skomplikowanych i długich formuł. Dodatkową zaletą używania megaformuł jest fakt, że działają one zwykle szybciej niż serie formuł, z których się składają.